
Abstract

ResClasses is a package forGAP 4, which provides a fully-featured and easy-to-use implementation of set-
theoretic unions of residue classes of the integers and of a few other rings.

The class of sets whichResClasses can deal with includes the open and the closed sets in the topology
on the respective ring which is induced by taking the set of all residue classes as a basis, as far as the usual
restrictions imposed by the finiteness of computing resources permit this.

The package further provides slightly more specialized functionality for unions of residue classes with
distinguished representatives and signed moduli.

TheResClasses package is used in a group theoretical context by theRCWA package [Koh07].

Copyright

c© 2003 - 2007 by Stefan Kohl
This package is distributed under the GNU General Public License.

Chapter 1

Set-Theoretic Unions of Residue Classes

1.1 Entering residue classes and set-theoretic unions thereof

1.1.1 ResidueClass (by ring, modulus and residue)

♦ ResidueClass(R, m, r ) (function)

♦ ResidueClass(m, r ) (function)

♦ ResidueClass(r, m ) (function)

Returns: In the three-argument form the residue classr mod mof the ringR, and in the two-
argument form the residue classr modmof the “default ring” (→ DefaultRing in theGAP Reference
Manual) of the arguments.

In the two-argument case,m is taken to be the larger andr is taken to be the smaller of the
arguments. For convenience, it is permitted to enclose the argument list in list brackets.

Residue classes have the propertyIsResidueClass. Rings are regarded as residue class
0 (mod 1), and therefore have this property. There are operationsModulus andResidue to retrieve
the modulusmresp. residuer of a residue class.

Example

gap> ResidueClass(2,3);
The residue class 2(3) of Z
gap> ResidueClass(Z_pi([2,5]),2,1);
The residue class 1(2) of Z_( 2, 5 )
gap> R := PolynomialRing(GF(2),1);;
gap> x := Indeterminate(GF(2),1);; SetName(x,"x");
gap> ResidueClass(R,x+One(R),Zero(R));
The residue class 0*Z(2) ( mod x+Z(2)ˆ0 ) of GF(2)[x]

1.1.2 ResidueClassUnion (by ring, modulus and residues)

♦ ResidueClassUnion(R, m, r ) (function)

♦ ResidueClassUnion(R, m, r, included, excluded ) (function)

Returns: The union of the residue classesr [i] mod m of the ring R, plus / minus finite sets
included andexcluded of elements ofR.

5

Example

gap> ResidueClassUnion(Integers,5,[1,2],[3,8],[-4,1]);
(Union of the residue classes 1(5) and 2(5) of Z) U [ 3, 8 ] \ [ -4, 1 ]
gap> ResidueClassUnion(Z_pi([2,3]),8,[3,5]);
Union of the residue classes 3(8) and 5(8) of Z_( 2, 3 )
gap> ResidueClassUnion(R,xˆ2,[One(R),x],[],[One(R)]);
<union of 2 residue classes (mod xˆ2) of GF(2)[x]> \ [ Z(2)ˆ0 ]

When talking about aresidue class unionin this chapter, we always mean an object as it is returned
by this function.

There are operationsModulus, Residues, IncludedElements andExcludedElements to re-
trieve the components of a residue class union as they have originally been passed as arguments to
ResidueClassUnion (1.1.2).

The user has the choice between a longer and more descriptive and a shorter and less bulky output
format for residue classes and unions thereof:

Example

gap> ResidueClassUnionViewingFormat("short");
gap> ResidueClassUnion(Integers,12,[0,1,4,7,8]);
0(4) U 1(6)
gap> ResidueClassUnionViewingFormat("long");
gap> ResidueClassUnion(Integers,12,[0,1,4,7,8]);
Union of the residue classes 0(4) and 1(6) of Z

1.1.3 AllResidueClassesModulo (of a given ring, modulo a given modulus)

♦ AllResidueClassesModulo(R, m) (function)

♦ AllResidueClassesModulo(m) (function)

Returns: A sorted list of all residue classes (modm) of the ringR.
If the argumentR is omitted it defaults to the default ring ofm – cf. the documentation of

DefaultRing in the GAP reference manual. A transversal for the residue classes (modm) can be
obtained by the operationAllResidues(R,m), and their number can be determined by the operation
NumberOfResidues(R,m).

Example

gap> AllResidueClassesModulo(Integers,2);
[ The residue class 0(2) of Z, The residue class 1(2) of Z ]
gap> AllResidueClassesModulo(Z_pi(2),2);
[ The residue class 0(2) of Z_( 2 ), The residue class 1(2) of Z_( 2 ) ]
gap> AllResidueClassesModulo(R,x);
[ The residue class 0*Z(2) ( mod x ) of GF(2)[x],
The residue class Z(2)ˆ0 ( mod x ) of GF(2)[x] ]

gap> AllResidues(R,xˆ3);
[ 0*Z(2), Z(2)ˆ0, x, x+Z(2)ˆ0, xˆ2, xˆ2+Z(2)ˆ0, xˆ2+x, xˆ2+x+Z(2)ˆ0 ]
gap> NumberOfResidues(Z_pi([2,3]),360);
72

1.2 Methods for residue class unions

There are methods forPrint, String andDisplay which are applicable to residue class unions.
There is a method forin which tests whether some ring element lies in a given residue class union.

Example

gap> Print(ResidueClass(1,2),"\n");
ResidueClassUnion( Integers, 2, [ 1 ] )
gap> 1 in ResidueClass(1,2);
true

There are methods forUnion, Intersection, Difference andIsSubset available for residue class
unions. They also accept finite subsets of the base ring as arguments.

Example

gap> S := Union(ResidueClass(0,2),ResidueClass(0,3));
Z \ Union of the residue classes 1(6) and 5(6) of Z
gap> Intersection(S,ResidueClass(0,7));
Union of the residue classes 0(14) and 21(42) of Z
gap> Difference(S,ResidueClass(2,4));
Union of the residue classes 0(4) and 3(6) of Z
gap> IsSubset(ResidueClass(0,2),ResidueClass(4,8));
true
gap> Union(S,[1..10]);
(Union of the residue classes 0(2) and 3(6) of Z) U [ 1, 5, 7 ]
gap> Intersection(S,[1..6]);
[ 2, 3, 4, 6 ]
gap> Difference(S,[1..6]);
(Union of the residue classes 0(2) and 3(6) of Z) \ [ 2, 3, 4, 6 ]
gap> Difference(Integers,[1..10]);
Z \ <set of cardinality 10>
gap> IsSubset(S,[1..10]);
false

If the underlying ring has a residue class ring of a given cardinalityt, then a residue class can be
written as a disjoint union oft residue classes with equal moduli:

1.2.1 SplittedClass (for a residue class and a number of parts)

♦ SplittedClass(cl, t ) (operation)

Returns: A partition of the residue classcl into t residue classes with equal moduli, provided
that such a partition exists. Otherwisefail.

Example

gap> SplittedClass(ResidueClass(1,2),2);
[ The residue class 1(4) of Z, The residue class 3(4) of Z ]
gap> SplittedClass(ResidueClass(Z_pi(3),3,0),2);
fail



ResClasses 8

Often one needs a partition of a given residue class union into “few” residue classes. The following
operation takes care of this:

1.2.2 AsUnionOfFewClasses (for a residue class union)

♦ AsUnionOfFewClasses(U) (operation)

Returns: A set of disjoint residue classes whose union is equal toU, up to the finite sets
IncludedElements(U) andExcludedElements(U).

As the name of the operation suggests, it is taken care that the number of residue classes in the
returned list is kept “reasonably small”. It is not guaranteed that it is minimal.

Example

gap> ResidueClassUnionViewingFormat("short");
gap> AsUnionOfFewClasses(Difference(Integers,ResidueClass(0,30)));
[ 1(2), 2(6), 4(6), 6(30), 12(30), 18(30), 24(30) ]
gap> Union(last);
Z \ 0(30)

One can compute the sets of sums, differences, products and quotients of the elements of a residue
class union and an element of the base ring:

Example

gap> ResidueClass(0,2) + 1;
1(2)
gap> ResidueClass(0,2) - 2 = ResidueClass(0,2);
true
gap> 3 * ResidueClass(0,2);
0(6)
gap> ResidueClass(0,2)/2;
Integers

1.2.3 PartitionsIntoResidueClasses (of a given ring, of given length)

♦ PartitionsIntoResidueClasses(R, length ) (operation)

Returns: A sorted list of all partitions of the ringR into length residue classes.

Example

gap> PartitionsIntoResidueClasses(Integers,4);
[ [ 0(2), 1(4), 3(8), 7(8) ], [ 0(2), 3(4), 1(8), 5(8) ],
[ 0(2), 1(6), 3(6), 5(6) ], [ 1(2), 0(4), 2(8), 6(8) ],
[ 1(2), 2(4), 0(8), 4(8) ], [ 1(2), 0(6), 2(6), 4(6) ],
[ 0(3), 1(3), 2(6), 5(6) ], [ 0(3), 2(3), 1(6), 4(6) ],
[ 1(3), 2(3), 0(6), 3(6) ], [ 0(4), 1(4), 2(4), 3(4) ] ]

1.2.4 RandomPartitionIntoResidueClasses (of a given ring, of given length)

♦ RandomPartitionIntoResidueClasses(R, length, primes ) (operation)

Returns: A “random” partition of the ringR into length residue classes whose moduli have
only prime factors inprimes , respectivelyfail if no such partition exists.

Example

gap> RandomPartitionIntoResidueClasses(Integers,30,[2,3,5,7]);
[ 0(7), 2(7), 5(7), 3(14), 10(14), 1(21), 8(21), 15(21), 18(21), 20(21),
6(63), 13(63), 25(63), 27(63), 32(63), 34(63), 46(63), 48(63), 53(63),
55(63), 4(126), 67(126), 137(189), 74(567), 200(567), 263(567),
389(567), 452(567), 11(1134), 578(1134) ]

gap> Union(last);
Integers
gap> Sum(List(last2,Density));
1

1.2.5 Density (of a residue class union)

♦ Density(U) (operation)

Returns: The natural density ofU as a subset of the underlying ring.
The natural densityof a residue classr(m) of a ring R is defined by 1/|R/mR|, and thenatural

densityof a unionU of finitely many residue classes is defined by the sum of the densities of the
elements of a partition ofU into finitely many residue classes.

Example

gap> Density(ResidueClass(0,2));
1/2
gap> Density(Difference(Integers,ResidueClass(0,5)));
4/5

For looping over residue class unions of the integers, there are methods for the operations
Iterator andNextIterator.

1.3 The categories and families of residue class unions

1.3.1 IsResidueClassUnion

♦ IsResidueClassUnion(U) (filter)

♦ IsResidueClassUnionOfZ(U) (filter)

♦ IsResidueClassUnionOfZ pi(U) (filter)

♦ IsResidueClassUnionOfGFqx(U) (filter)

Returns: true if U is a residue class union, a residue class union of the ring of integers, a residue
class union of a semilocalization of the ring of integers or a residue class union of a polynomial ring
in one variable over a finite field, respectively, andfalse otherwise.

Often the same methods can be used for residue class unions of the ring of integers and of its
semilocalizations. For this reason, there is a categoryIsResidueClassUnionOfZorZ pi which is

the union ofIsResidueClassUnionOfZ andIsResidueClassUnionOfZ pi. The internal represen-
tation of residue class unions is calledIsResidueClassUnionResidueListRep. There are methods
available forExtRepOfObj andObjByExtRep.

1.3.2 ResidueClassUnionsFamily (of a ring)

♦ ResidueClassUnionsFamily(R) (function)

♦ ResidueClassUnionsFamily(R, fixedreps ) (function)

Returns: The family of residue class unions or the family of unions of residue classes with fixed
representatives of the ringR, depending on whetherfixedreps is present andtrue or not.

The ringRcan be retrieved asUnderlyingRing(ResidueClassUnionsFamily(R)). There is no
coercion between residue class unions or unions of residue classes with fixed representatives which
belong to different families. Unions of residue classes with fixed representatives are described in the
next chapter.



Chapter 2

Unions of Residue Classes with Fixed
Representatives

ResClasses supports computations with unions of residue classes which are endowed with distin-
guished (“fixed”) representatives. These unions of residue classes can be viewed as multisets of ring
elements. The residue classes forming such a union do not need to be disjoint or even only distinct.

2.1 Entering unions of residue classes with fixed representatives

2.1.1 ResidueClassWithFixedRepresentative (by ring, modulus and residue)

♦ ResidueClassWithFixedRepresentative(R, m, r ) (function)

♦ ResidueClassWithFixedRepresentative(m, r ) (function)

Returns: The residue classr modmof the ringR, with the fixed representativer .
If the argumentR is omitted, it defaults toIntegers. Residue classes with fixed representatives

have the propertyIsResidueClassWithFixedRepresentative. The fixed representativer can be
retrieved by the operationResidue, and the modulusmcan be retrieved by the operationModulus.

Example

gap> ResidueClassWithFixedRepresentative(Integers,2,1);
[1/2]

2.1.2 UnionOfResidueClassesWithFixedReps (by ring and list of classes)

♦ UnionOfResidueClassesWithFixedReps(R, classes ) (function)

♦ UnionOfResidueClassesWithFixedReps(classes ) (function)

Returns: The union of the residue classesclasses [i][2] mod classes [i][1] of the ring R,
with fixed representativesclasses [i][2].

The argumentclasses must be a list of pairs of elements of the ringR. Their first entries – the
moduli – must be nonzero. If the argumentR is omitted, it defaults toIntegers.

11

Example

gap> UnionOfResidueClassesWithFixedReps(Integers,[[2,4],[3,9]]);
[4/2] U [9/3]

There is a method for the operationModulus which returns the lcm of the moduli of the residue
classes forming such a union. Further there is an operationClasses for retrieving the list of classes
which has been passed as an argument toUnionOfResidueClassesWithFixedReps. The operation
AsListOfClasses does the same, except that the returned list contains residue classes instead of pairs
[modulus ,residue ]. There are methods forPrint, String andDisplay available for unions of
residue classes with fixed representatives.

2.1.3 AllResidueClassesWithFixedRepsModulo (by ring and modulus)

♦ AllResidueClassesWithFixedRepsModulo(R, m) (function)

♦ AllResidueClassesWithFixedRepsModulo(m) (function)

Returns: A sorted list of all residue classes (modm) of the ringR, with fixed representatives.
If the argumentR is omitted it defaults to the default ring ofm, cf. the documentation of

DefaultRing in the GAP reference manual. The representatives are the same as those chosen by
the operationmod. See alsoAllResidueClassesModulo (1.1.3).

Example

gap> AllResidueClassesWithFixedRepsModulo(4);
[ [0/4], [1/4], [2/4], [3/4] ]

2.2 Methods for unions of residue classes with fixed representatives

Throughout this chapter, the argumentRdenotes the underlying ring, and the argumentsU, U1andU2
denote unions of residue classes ofRwith fixed representatives.

Unions of residue classes with fixed representatives are multisets. Elements and residue classes
can be contained with multiplicities:

2.2.1 Multiplicity (of an element in a union of residue classes with fixed rep’s)

♦ Multiplicity(x, U ) (method)

♦ Multiplicity(cl, U ) (method)

Returns: The multiplicity ofx in U regarded as a multiset of ring elements, resp. the multiplicity
of the residue classcl in U regarded as a multiset of residue classes.

Example

gap> U := UnionOfResidueClassesWithFixedReps(Integers,[[2,0],[3,0]]);
[0/2] U [0/3]
gap> List([0..20],n->Multiplicity(n,U));
[ 2, 0, 1, 1, 1, 0, 2, 0, 1, 1, 1, 0, 2, 0, 1, 1, 1, 0, 2, 0, 1 ]
gap> Multiplicity(ResidueClassWithFixedRep(2,0),U);
1

Let U be a union of residue classes with fixed representatives. The multisetU can have an attribute
Density which denotes itsnatural densityas a multiset, i.e. elements with multiplicityk countk-
fold. The multisetU has the propertyIsOverlappingFree if it consists of pairwise disjoint residue
classes. The set-theoretic union of the residue classes formingU can be determined by the operation
AsOrdinaryUnionOfResidueClasses. The object returned by this operation is an “ordinary” residue
class union as described in Chapter1.

Example

gap> U := UnionOfResidueClassesWithFixedReps(Integers,[[2,0],[3,0]]);
[0/2] U [0/3]
gap> Density(U);
5/6
gap> IsOverlappingFree(U);
false
gap> AsOrdinaryUnionOfResidueClasses(U);
Z \ Union of the residue classes 1(6) and 5(6) of Z
gap> Density(last);
2/3

In the sequel we abbreviate the term “the multiset of ring elements endowed with the structure of a
union of residue classes with fixed representatives” by “the multiset”.

There are methods for+ and- available for computing the multiset of sumsu+ x, u ∈ U , the
multiset of differencesu−x resp.x−u, u∈U and the multiset of the additive inverses of the elements
of U . Further there are methods for* and/ available for computing the multiset of productsx · u,
u∈U and the multiset of quotientsu/x, u∈U . The division method requires all elements ofU to be
divisible byx. If the underlying ring is the ring of integers, scalar multiplication and division leaveδ
invariant (→ Delta (2.3.1)).

Example

gap> U := UnionOfResidueClassesWithFixedReps(Integers,[[2,0],[3,0]]);
[0/2] U [0/3]
gap> U + 7;
[7/2] U [7/3]
gap> U - 7; 7 - U; -U;
[-7/2] U [-7/3]
[7/-3] U [7/-2]
[0/-3] U [0/-2]
gap> V := 2 * U;
[0/4] U [0/6]
gap> V/2;
[0/2] U [0/3]

2.2.2 Union (for unions of residue classes with fixed representatives)

♦ Union(U1, U2 ) (method)

Returns: The union ofU1 andU2.
The multiplicity of any ring element or residue class in the union is the sum of its multiplicities

in the arguments. It holds thatDelta(Union(U1,U2)) = Delta(U1) + Delta(U2). (→ Delta
(2.3.1)).

Example

gap> U := UnionOfResidueClassesWithFixedReps(Integers,[[2,0],[3,0]]);
[0/2] U [0/3]
gap> Union(U,U);
[0/2] U [0/2] U [0/3] U [0/3]

2.2.3 Intersection (for unions of residue classes with fixed representatives)

♦ Intersection(U1, U2 ) (method)

Returns: The intersection ofU1 andU2.
The multiplicity of any residue class in the intersection is the minimum of its multiplicities in the

arguments.
Example

gap> U := UnionOfResidueClassesWithFixedReps(Integers,[[2,0],[3,0]]);
[0/2] U [0/3]
gap> Intersection(U,ResidueClassWithFixedRep(2,0));
[0/2]
gap> Intersection(U,ResidueClassWithFixedRep(6,0));
Empty union of residue classes of Z with fixed representatives

2.2.4 Difference (for unions of residue classes with fixed representatives)

♦ Difference(U1, U2 ) (method)

Returns: The difference ofU1 andU2.
The multiplicity of any residue class in the difference is its multiplicity inU1 minus its mul-

tiplicity in U2, if this value is nonnegative. The difference of the empty residue class union
with fixed representatives and some residue class[r/m] is set equal to[(m− r)/m]. It holds that
Delta(Difference(U1,U2)) = Delta(U1) - Delta(U2). (→ Delta (2.3.1)).

Example

gap> U := UnionOfResidueClassesWithFixedReps(Integers,[[2,0],[3,0]]);
[0/2] U [0/3]
gap> V := UnionOfResidueClassesWithFixedReps(Integers,[[3,0],[5,2]]);
[0/3] U [2/5]
gap> Difference(U,V);
[0/2] U [3/5]

2.3 The invariant Delta

2.3.1 Delta (for a union of residue classes with fixed representatives)

♦ Delta(U) (attribute)

Returns: The value of the invariantδ of the residue class unionU.
For a residue class[r/m] with fixed representative we setδ([r/m]) := r/m−1/2, and extend this

definition additively to unions of such residue classes. If no representatives are fixed, this definition
is still unique (mod 1). There is a related invariantρ which is defined byeδ(U)πi . The corresponding
attribute is calledRho.

Example

gap> U := UnionOfResidueClassesWithFixedReps(Integers,[[2,3],[3,4]]);
[3/2] U [4/3]
gap> Delta(U) = (3/2-1/2) + (4/3-1/2);
true
gap> V := RepresentativeStabilizingRefinement(U,3);
[3/6] U [5/6] U [7/6] U [4/9] U [7/9] U [10/9]
gap> Delta(V) = Delta(U);
true
gap> Rho(V);
E(12)ˆ11

2.3.2 RepresentativeStabilizingRefinement (of a union of res.-classes with fixed rep’s)

♦ RepresentativeStabilizingRefinement(U, k ) (method)

Returns: The representative stabilizing refinement ofU into k parts.
The representative stabilizing refinementof a residue class[r/m] of Z into k parts is defined by

[r/km]∪ [(r + m)/km]∪ . . .∪ [(r +(k−1)m)/km]. This definition is extended in the obvious way to
unions of residue classes.

If the argumentk is zero, the method performs a simplification ofU by joining appropriate residue
classes, if this is possible.

In any case the value ofDelta(U) is invariant under this operation (→ Delta (2.3.1)).
Example

gap> U := UnionOfResidueClassesWithFixedReps(Integers,[[2,0],[3,0]]);
[0/2] U [0/3]
gap> RepresentativeStabilizingRefinement(U,4);
[0/8] U [2/8] U [4/8] U [6/8] U [0/12] U [3/12] U [6/12] U [9/12]
gap> RepresentativeStabilizingRefinement(last,0);
[0/2] U [0/3]

2.4 The categories of unions of residue classes with fixed rep’s

The names of the categories of unions of residue classes with fixed represen-
tatives are IsUnionOfResidueClasses[ OfZ | OfZ pi | OfZorZ pi | OfGFqx
]WithFixedRepresentatives.



Chapter 3

Semilocalizations of the Integers

This package implements residue class unions of the semilocalizationsZ(π) of the ring of integers. It
also provides the underlyingGAP implementation of these rings themselves.

3.1 Entering semilocalizations of the integers

3.1.1 Z pi (by set of noninvertible primes)

♦ Z pi(pi ) (function)

♦ Z pi(p) (function)

Returns: The ringZ(π) or the ringZ(p), respectively.
The returned ring has the propertyIsZ pi. The setpi of noninvertible primes can be retrieved by

the operationNoninvertiblePrimes.
Example

gap> R := Z_pi(2);
Z_( 2 )
gap> S := Z_pi([2,5,7]);
Z_( 2, 5, 7 )

3.2 Methods for semilocalizations of the integers

There are methods for the operationsin, Intersection, IsSubset, StandardAssociate, Gcd, Lcm,
Factors andIsUnit available for semilocalizations of the integers. For the documentation of these
operations, see theGAP reference manual. The standard associate of an element of a ringZ(π) is
defined by the product of the noninvertible prime factors of its numerator.

Example

gap> 4/7 in R; 3/2 in R;
true
false
gap> Intersection(R,Z_pi([3,11])); IsSubset(R,S);
Z_( 2, 3, 11 )
true

Example

gap> StandardAssociate(R,-6/7);
2
gap> Gcd(S,90/3,60/17,120/33);
10
gap> Lcm(S,90/3,60/17,120/33);
40
gap> Factors(R,840);
[ 105, 2, 2, 2 ]
gap> Factors(R,-2/3);
[ -1/3, 2 ]
gap> IsUnit(S,3/11);
true



Chapter 4

Installation and auxiliary functions

4.1 Requirements

TheResClasses package needs at leastGAP 4.4.7 andGAPDoc 0.99999 [LN07]. It can be used under
UNIX, under Windows and on the MacIntosh.ResClasses is completely written in theGAP language
and does neither contain nor require external binaries.

4.2 Installation

Like any otherGAP package,ResClasses must be installed in thepkg subdirectory of theGAP dis-
tribution. This is accomplished by extracting the distribution file in this directory. By default, the
packageResClasses is autoloaded. If you have switched autoloading of packages off, you can load
ResClasses via LoadPackage( "resclasses" );.

4.3 The testing routine

4.3.1 ResClassesTest

♦ ResClassesTest() (function)

Returns: Nothing.
Performs tests of theResClasses package. Errors, i.e. differences to the correct results of the test

computations, are reported. The processed test files are in the directorypkg/resclasses/tst.

4.4 Building the manual

4.4.1 ResClassesBuildManual

♦ ResClassesBuildManual() (function)

Returns: Nothing.
This function is a development tool which builds the manual of theResClasses package in the

file formats LATEX, PDF, HTML and ASCII text. This is accomplished using theGAPDoc package
by Frank L̈ubeck and Max Neunḧoffer. Building the manual is possible only on UNIX systems and
requires PDFLATEX. As all files generated by this function are included in the distribution file anyway,
users will not need it.

18

4.5 SendEmail and EmailLogFile

4.5.1 SendEmail

♦ SendEmail(sendto, copyto, subject, text ) (function)

Returns: Zero if everything worked correctly, and a system error number otherwise.
Sends an e-mail with subjectsubject and bodytext to the addresses in the listsendto , and

copies it to those in the listcopyto . The first two arguments must be lists of strings, and the latter
two must be strings.

As most system-related functions,SendEmail works only under UNIX. Further, the computer
must of course be connected to the internet.

4.5.2 EmailLogFile

♦ EmailLogFile(addresses ) (function)

Returns: Zero if everything worked correctly, and a system error number otherwise.
Sends the current log file by e-mail toaddresses , if GAP is in logging mode and one is working

under UNIX, and does nothing otherwise. The argumentaddresses must be either a list of e-mail
addresses or a single e-mail address. Long log files are abbreviated, i.e. if the log file is larger than
64KB, then any output is truncated at 1KB, and if the log file is still longer than 64KB afterwards, it
is truncated at 64KB.

