
Chapter 1

Introduction

This is the manual for theGAP packageQuaGroup, for doing computations with quantized enveloping
algebras of semisimple Lie algebras.

Apart from the chapter you are currently reading, this document consists of two chapters. In
Chapter2 we give a short summary of parts of the theory of quantized enveloping algebras. This fixes
the notations and definitions that we use. Then in Chapter3 we describe the functions that constitute
the package.

The package can be obtained fromhttp://www.math.uu.nl/people/graaf/quagroup.html
The directoryquagroup/doc contains the manual of the package indvi, ps, pdf and html for-
mat. The manual was built with theGAP share packageGAPDoc, [LN01]. This means that, in
order to be able to use the on-line help ofQuaGroup, you have to installGAPDoc before calling
LoadPackage("quagroup");.

The main algorithm of the package (on which virtually the whole functionality relies) is a method
for computing with so-called PBW-type bases, analogous to Poincaré-Birkhoff-Witt bases in universal
enveloping algebras. In both cases commutation relations between the generators are used. However,
in the latter case all commutation relations are of the formyx= xy+z, wherex,y are generators, and
z is a linear combination of generators. In the case of quantized enveloping algebras the situation is
generally much more complicated. For example, in the quantized enveloping algebra of typeE7 we
have the following relation:

Example
F62*F26 = (q)*F26*F62+(1-qˆ2)*F28*F61+(-q+qˆ3)*F30*F60+(qˆ2-qˆ4)*F31*F59+

(qˆ2-qˆ4)*F33*F58+(-qˆ3+qˆ5)*F34*F57+(qˆ4-qˆ6)*F35*F56+
(qˆ-1-q-qˆ5+qˆ7)*F36*F55+(qˆ6)*F54

Due to the complexity of these commutation relations, some computations (even with rather small
input) may take quite some time.

Remark: The package can deal with quantized enveloping algebras corresponding to root systems
of rank at least up to eight, exceptE8. In that case the computation of the necessary commutation re-
lations took more than 2 GB. I wish to thank Steve Linton for trying this computation on the machines
in St Andrews.

The following example illustrates some of the features of the package.
Example

# We define a root system by giving its type:
gap> R:= RootSystem( "B", 2 );
<root system of type B2>
# Corresponding to the root system we define a quantized enveloping algebra:

6

gap> U:= QuantizedUEA( R );
QuantumUEA( <root system of type B2>, Qpar = q )
# It is generated by the generators of a so-called PBW-type basis:
gap> GeneratorsOfAlgebra( U );
[ F1, F2, F3, F4, K1, K1+(qˆ-2-qˆ2)*[ K1 ; 1 ], K2, K2+(qˆ-1-q)*[ K2 ; 1 ],

E1, E2, E3, E4 ]
# We can construct highest-weight modules:
gap> V:= HighestWeightModule( U, [1,1] );
<16-dimensional left-module over QuantumUEA( <root system of type B
2>, Qpar = q )>
# For modules of small dimension we can compute the corresponding
# R-matrix:
gap> U:= QuantizedUEA( RootSystem("A",2) );;
gap> V:= HighestWeightModule( U, [1,0] );;
gap> RMatrix( V );
[ [ qˆ2, 0, 0, 0, 0, 0, 0, 0, 0 ], [ 0, qˆ3, 0, qˆ2-qˆ4, 0, 0, 0, 0, 0 ],

[ 0, 0, qˆ3, 0, 0, 0, qˆ2-qˆ4, 0, 0 ], [ 0, 0, 0, qˆ3, 0, 0, 0, 0, 0 ],
[ 0, 0, 0, 0, qˆ2, 0, 0, 0, 0 ], [ 0, 0, 0, 0, 0, qˆ3, 0, qˆ2-qˆ4, 0 ],
[ 0, 0, 0, 0, 0, 0, qˆ3, 0, 0 ], [ 0, 0, 0, 0, 0, 0, 0, qˆ3, 0 ],
[ 0, 0, 0, 0, 0, 0, 0, 0, qˆ2 ] ]

# We can compute elements of the canonical basis of the "negative" part
# of a quantized enveloping algebra:
gap> U:= QuantizedUEA( RootSystem("F",4) );;
gap> B:= CanonicalBasis( U );
<canonical basis of QuantumUEA( <root system of type F4>, Qpar = q ) >
gap> p:= PBWElements( B, [0,1,2,1] );
[ F3*F9ˆ(2)*F24, F3*F9*F23+(qˆ2)*F3*F9ˆ(2)*F24,

(q+qˆ3)*F3*F9ˆ(2)*F24+F7*F9*F24, (qˆ2)*F3*F9*F23+(qˆ2+qˆ4)*F3*F9ˆ(2)*F
24+(q)*F7*F9*F24+F7*F23, (qˆ4)*F3*F9ˆ(2)*F24+(q)*F7*F9*F24+F8*F24,

(qˆ4)*F3*F9*F23+(qˆ6)*F3*F9ˆ(2)*F24+(qˆ3)*F7*F9*F24+(qˆ2)*F7*F23+(qˆ2)*F
8*F24+F9*F21, (q+qˆ3)*F3*F9*F23+(qˆ3+qˆ5)*F3*F9ˆ(2)*F24+(qˆ2)*F7*F9*F
24+(q)*F7*F23+(q)*F9*F21+F16 ]

# We can construct (anti-) automorphisms of quantized enveloping
# algebras:
gap> t:= AntiAutomorphismTau( U );
<anti-automorphism of QuantumUEA( <root system of type F4>, Qpar = q )>
gap> Image( t, p[1] );
(qˆ4)*F3*F9*F23+(qˆ6)*F3*F9ˆ(2)*F24+(qˆ3)*F7*F9*F24+(qˆ2)*F7*F23+(qˆ2)*F8*F
24+F9*F21
# (This is the sixth element of p.)



Chapter 2

Background

In this chapter we summarize some of the theoretical concepts with whichQuaGroup operates. Due
to the rather mathematical nature of this chapter everything has been written in LaTeX. Therefore, it
will be almost unreadable in the html version.

2.1 Gaussian Binomials

Let v be an indeterminate overQ. For a positive integern we set

[n] = vn−1 +vn−3 + · · ·+v−n+3 +v−n+1.

We say that[n] is the Gaussian integercorresponding ton. The Gaussian factorial[n]! is defined
by

[0]! = 1, [n]! = [n][n−1] · · · [1], for n > 0.

Finally, the Gaussian binomialis [
n
k

]
=

[n]!
[k]![n−k]!

.

2.2 Quantized enveloping algebras

Let g be a semisimple Lie algebra with root systemΦ. By ∆ = {α1, . . . ,αl} we denote a fixed simple
system ofΦ. Let C = (Ci j ) be the Cartan matrix ofΦ (with respect to∆, i.e.,Ci j = 〈αi ,α∨

j 〉). Let
d1, . . . ,dl be the unique sequence of positive integers with greatest common divisor 1, such thatdiCji =
d jCi j , and set(αi ,α j) = d jCi j . (We note that this implies that(αi ,αi) is divisible by 2.) ByPwe denote
the weight lattice, and we extend the form( , ) to P by bilinearity.

By W(Φ) we denote the Weyl group ofΦ. It is generated by the simple reflectionssi = sαi for
1≤ i ≤ l (wheresα is defined bysα(β) = β−〈β,α∨〉α).

We work over the fieldQ(q). Forα ∈ Φ we set

qα = q
(α,α)

2 ,

and for a non-negative integern, [n]α = [n]v=qα ; [n]α! and

[
n
k

]
α

are defined analogously.

8



QuaGroup 9

The quantized enveloping algebraUq(g) is the associative algebra (with one) overQ(q) generated
by Fα, Kα, K−1

α , Eα for α ∈ ∆, subject to the following relations

KαK−1
α = K−1

α Kα = 1, KαKβ = KβKα

EβKα = q−(α,β)KαEβ

KαFβ = q−(α,β)FβKα

EαFβ = FβEα +δα,β
Kα−K−1

α

qα−q−1
α

together with, forα 6= β ∈ ∆,

1−〈β,α∨〉

∑
k=0

(−1)k
[
1−〈β,α∨〉

k

]
α

E1−〈β,α∨〉−k
α EβEk

α = 0

1−〈β,α∨〉

∑
k=0

(−1)k
[
1−〈β,α∨〉

k

]
α

F1−〈β,α∨〉−k
α FβFk

α = 0.

The quantized enveloping algebra has an automorphismω defined byω(Fα) = Eα, ω(Eα) = Fα and
ω(Kα) = K−1

α . Also there is an anti-automorphismτ defined byτ(Fα) = Fα, τ(Eα) = Eα andτ(Kα) =
K−1

α . We haveω2 = 1 andτ2 = 1.
If the Dynkin diagram ofΦ admits a diagram automorphismπ, thenπ induces an automorphism

of Uq(g) in the obvious way (π is a permutation of the simple roots; we permute theFα, Eα, K±1
α

accordingly).
Now we viewUq(g) as an algebra overQ, and we let : Uq(g) →Uq(g) be the automorphism

defined byFα = Fα, Kα = K−1
α , Eα = Eα, q = q−1.

2.3 Representations ofUq(g)

Let λ ∈P be a dominant weight. Then there is a unique irreducible highest-weight module overUq(g)
with highest weightλ. We denote it byV(λ). It has the same character as the irreducible highest-
weight module overg with highest weightλ. Furthermore, every finite-dimensionalUq(g)-module is
a direct sum of irreducible highest-weight modules.

It is well-known thatUq(g) is a Hopf algebra. The comultiplication∆ : Uq(g)→Uq(g)⊗Uq(g) is
defined by

∆(Eα) = Eα⊗1+Kα⊗Eα

∆(Fα) = Fα⊗K−1
α +1⊗Fα

∆(Kα) = Kα⊗Kα.

(Note that we use the same symbol to denote a simple system ofΦ; of course this does not cause
confusion.) The counitε : Uq(g)→Q(q) is a homomorphism defined byε(Eα) = ε(Fα) = 0, ε(Kα) =
1. Finally, the antipodeS : Uq(g) → Uq(g) is an anti-automorphism given byS(Eα) = −K−1

α Eα,
S(Fα) =−FαKα, S(Kα) = K−1

α .
Using∆ we can make the tensor productV⊗W of twoUq(g)-modulesV,W into aUq(g)-module.

The counitε yields a trivial 1-dimensionalUq(g)-module. And withSwe can define aUq(g)-module
structure on the dualV∗ of aUq(g)-moduleV, by (u· f )(v) = f (S(u) ·v).

The Hopf algebra structure given above is not the only one possible. For example, we can twist
∆,ε,Sby an automorphism, or an anti-automorphismf . The twisted comultiplication is given by

∆ f = f ⊗ f ◦∆◦ f−1.

The twisted antipode by

Sf =

{
f ◦S◦ f−1 if f is an automorphism

f ◦S−1◦ f−1 if f is an anti-automorphism.

And the twisted counit byε f = ε◦ f−1 (see [Jan96], 3.8).

2.4 PBW-type bases

The first problem one has to deal with when working withUq(g) is finding a basis of it, along with an
algorithm for expressing the product of two basis elements as a linear combination of basis elements.
First of all we have thatUq(g) ∼= U−⊗U0⊗U+ (as vector spaces), whereU− is the subalgebra
generated by theFα, U0 is the subalgebra generated by theKα, andU+ is generated by theEα. So
a basis ofUq(g) is formed by all elementsFKE, whereF , K, E run through bases ofU−, U0, U+

respectively.
Finding a basis ofU0 is easy: it is spanned by allKr1

α1 · · ·K
r l
αl , wherer i ∈ Z. ForU−, U+ we use

the so-calledPBW-typebases. They are defined as follows. Forα,β∈ ∆ we setrβ,α =−〈β,α∨〉. Then
for α ∈ ∆ we have the automorphismTα : Uq(g)→Uq(g) defined by

Tα(Eα) =−FαKα

Tα(Eβ) =
rβ,α

∑
i=0

(−1)iq−i
α E

(rβ,α−i)
α EβE(i)

α (for α 6= β)

Tα(Kβ) = KβK
rβ,α
α

Tα(Fα) =−K−1
α Eα

Tα(Fβ) =
rβ,α

∑
i=0

(−1)iqi
αF(i)

α FβF
(rβ,α−i)

α (for α 6= β),

(whereE(k)
α = Ek

α/[k]α!, and likewise forF(k)
α ).

Let w0 = si1 · · ·sit be a reduced expression for the longest element in the Weyl groupW(Φ). For
1 ≤ k ≤ t setFk = Tαi1

· · ·Tαik−1
(Fαik

), andEk = Tαi1
· · ·Tαik−1

(Eαik
). ThenFk ∈ U−, andEk ∈ U+.

Furthermore, the elementsFm1
1 · · ·Fmt

t , En1
1 · · ·Ent

t (where themi , ni are non-negative integers) form
bases ofU− andU+ respectively.

The elementsFα andEα are said to have weight−α andα respectively, whereα is a simple root.
Furthermore, the weight of a productab is the sum of the weights ofa and b. Now elements of
U−, U+ that are linear combinations of elements of the same weight are said to be homogeneous. It
can be shown that the elementsFk, andEk are homogeneous of weight−β andβ respectively, where
β = si1 · · ·sik−1(αik).

In the sequel we use the notationF(m)
k = Fm

k /[m]αik
!, andE(n)

k = En
k/[n]αik

!.

2.5 The Z-form of Uq(g)

For α ∈ ∆ set [
Kα
n

]
=

n

∏
i=1

q−i+1
α Kα−qi−1

α K−1
α

qi
α−q−i

α
.

Then according to [Lus90], Theorem 6.7 the elements

F(k1)
1 · · ·F(kt)

t Kδ1
α1

[
Kα1

m1

]
· · ·Kδl

αl

[
Kαl

ml

]
E(n1)

1 · · ·E(nt)
t ,

(whereki ,mi ,ni ≥ 0, δi = 0,1) form a basis ofUq(g), such that the product of any two basis elements
is a linear combination of basis elements with coefficients inZ[q,q−1]. The quantized enveloping
algebra overZ[q,q−1] with this basis is called theZ-form of Uq(g), and denoted byUZ. SinceUZ is
defined overZ[q,q−1] we can specializeq to any nonzero elementε of a fieldF , and obtain an algebra
Uε overF .

We callq∈Q(q), andε ∈ F the quantum parameter ofUq(g) andUε respectively.
Let λ be a dominant weight, andV(λ) the irreducible highest weight module of highest weight

λ overUq(g). Let vλ ∈ V(λ) be a fixed highest weight vector. ThenUZ · vλ is aUZ-module. So by
specializingq to an elementε of a fieldF , we get aUε-module. We call it the Weyl module of highest
weightλ overUε. We note that it is not necessarily irreducible.

2.6 The canonical basis

As in Section2.4 we letU− be the subalgebra ofUq(g) generated by theFα for α ∈ ∆. In [Lus0a]
Lusztig introduced a basis ofU− with very nice properties, called thecanonical basis. (Later this basis
was also constructed by Kashiwara, using a different method. For a brief overview on the history of
canonical bases we refer to [Com06].)

Let w0 = si1 · · ·sit , and the elementsFk be as in Section2.4. Then, in order to stress the dependency
of the monomial

F(n1)
1 · · ·F(nt)

t (2.1)

on the choice of reduced expression for the longest element inW(Φ) we say that it is aw0-monomial.
Now we let be the automorphism ofU− defined in Section2.2. Elements that are invariant under

are said to be bar-invariant.
By results of Lusztig ([Lus93] Theorem 42.1.10, [Lus96], Proposition 8.2), there is a unique basis

B of U− with the following properties. Firstly, all elements ofB are bar-invariant. Secondly, for any
choice of reduced expressionw0 for the longest element in the Weyl group, and any elementX ∈ B
we have thatX = x+ ∑ζixi , wherex,xi arew0-monomials,x 6= xi for all i, andζi ∈ qZ[q]. The basis
B is called the canonical basis. If we work with a fixed reduced expression for the longest element in
W(Φ), and writeX ∈ B as above, then we say thatx is theprincipal monomialof X.

Let L be theZ[q]-lattice inU− spanned byB. ThenL is also spanned by allw0-monomials (where
w0 is a fixed reduced expression for the longest element inW(Φ)). Now let w̃0 be a second reduced
expression for the longest element inW(Φ). Let x be aw0-monomial, and letX be the element of
B with principal monomialx. Write X as a linear combination of̃w0-monomials, and let̃x be the
principal monomial of that expression. Then we writex̃ = Rw̃0

w0
(x). Note thatx = x̃ modqL .

Now let B be the set of allw0-monomials modqL . ThenB is a basis of theZ-moduleL/qL .
Moreover,B is independent of the choice ofw0. Let α ∈ ∆, and letw̃0 be a reduced expression for

the longest element inW(Φ), starting withsα. The Kashiwara operators̃Fα : B → B andẼα : B →
B ∪{0} are defined as follows. Letb ∈ B and letx be thew0-monomial such thatb = x modqL .
Set x̃ = Rw̃0

w0
(x). Thenx̃′ is thew̃0-monomial constructed from̃x by increasing its first exponent by

1 (the first exponent is then1 in (2.1)). ThenF̃α(b) = Rw0
w̃0

(x̃′) modqL . For Ẽα we let x̃′ be thew̃0-
monomial constructed from̃x by decreasing its first exponent by 1, if this exponent is≥ 1. Then
Ẽα(b) = Rw0

w̃0
(x̃′) modqL . Furthermore,̃Eα(b) = 0 if the first exponent of̃x is 0. It can be shown

that this definition does not depend on the choice ofw0, w̃0. Furthermore we havẽFαẼα(b) = b, if
Ẽα(b) 6= 0, andẼαF̃α(b) = b for all b∈ B.

Let w0 = si1 · · ·sit be a fixed reduced expression for the longest element inW(Φ). For b∈ B we
define a sequence of elementsbk ∈ B for 0≤ k≤ t, and a sequence of integersnk for 1≤ k≤ t as
follows. We setb0 = b, and ifbk−1 is defined we letnk be maximal such that̃Enk

αik
(bk−1) 6= 0. Also we

setbk = Ẽnk
αik

(bk−1). Then the sequence(n1, . . . ,nt) is called thestring of b∈ B (relative tow0). We

note thatb = F̃n1
αi1
· · · F̃nt

αit
(1). The set of all strings parametrizes the elements ofB, and hence ofB.

Now let V(λ) be a highest-weight module overUq(g), with highest weightλ. Let vλ be a fixed
highest weight vector. ThenBλ = {X · vλ | X ∈ B} \ {0} is a basis ofV(λ), called thecanonical
basis, or crystal basisof V(λ). Let L(λ) be theZ[q]-lattice inV(λ) spanned byBλ. We letB(λ) be
the set of allx · vλ modqL(λ), wherex runs through allw0-monomials, such thatX · vλ 6= 0, where
X ∈ B is the element with principal monomialx. Then the Kashiwara operators are also viewed as
mapsB(λ) → B(λ)∪{0}, in the following way. Letb = x · vλ modqL(λ) be an element ofB(λ),
and letb′ = x modqL be the corresponding element ofB. Let y be thew0-monomial such that
F̃α(b′) = y modqL . ThenF̃α(b) = y·vλ modqL(λ). The description of̃Eα is analogous. (In [Jan96],
Chapter 9 a different definition is given; however, by [Jan96], Proposition 10.9, Lemma 10.13, the
two definitions agree).

The setB(λ) has dimV(λ) elements. We letΓ be the coloured directed graph defined as follows.
The points ofΓ are the elements ofB(λ), and there is an arrow with colourα∈ ∆ connectingb,b′ ∈B,
if F̃α(b) = b′. The graphΓ is called thecrystal graphof V(λ).

2.7 The path model

In this section we recall some basic facts on Littelmann’s path model.
From Section2.2 we recall thatP denotes the weight lattice. LetPR be the vector space overR

spanned byP. Let Π be the set of all piecewise linear pathsξ : [0,1] → PR, such thatξ(0) = 0. For
α ∈ ∆ Littelmann defined operatorsfα,eα : Π → Π∪{0}. Let λ be a dominant weight and letξλ be
the path joiningλ and the origin by a straight line. LetΠλ be the set of all nonzerofαi1

· · · fαim
(ξλ) for

m≥ 0. Thenξ(1)∈P for all ξ∈Πλ. Letµ∈P be a weight, and letV(λ) be the highest-weight module
overUq(g) of highest weightλ. A theorem of Littelmann states that the number of pathsξ ∈ Πλ such
that ξ(1) = µ is equal to the dimension of the weight space of weightµ in V(λ) ([Lit95], Theorem
9.1).

All paths appearing inΠλ are so-called Lakshmibai-Seshadri paths (LS-paths for short). They are
defined as follows. Let≤ denote the Bruhat order onW(Φ). Forµ,ν ∈W(Φ) ·λ (the orbit ofλ under
the action ofW(Φ)), write µ≤ ν if τ ≤ σ, whereτ,σ ∈W(Φ) are the unique elements of minimal
length such thatτ(λ) = µ, σ(λ) = ν. Now a rational path of shapeλ is a pairπ = (ν,a), whereν =
(ν1, . . . ,νs) is a sequence of elements ofW(Φ) ·λ, such thatνi > νi+1 anda= (a0 = 0,a1, · · · ,as = 1)
is a sequence of rationals such thatai < ai+1. The pathπ corresponding to these sequences is given

by

π(t) =
r−1

∑
j=1

(a j −a j−1)ν j +νr(t−ar−1)

for ar−1≤ t ≤ ar . Now an LS-path of shapeλ is a rational path satisfying a certain integrality condition
(see [Lit94], [Lit95]). We note that the pathξλ = ((λ),(0,1)) joining the origin andλ by a straight
line is an LS-path.

Now from [Lit94], [Lit95] we transcribe the following:

• Let π be an LS-path. Thenfαπ is an LS-path or 0; and the same holds foreαπ.

• The action offα,eα can easily be described combinatorially (see [Lit94]).

• The endpoint of an LS-path is an integral weight.

• Let π = (ν,a) be an LS-path. Then byφ(π) we denote the unique elementσ of W(Φ) of shortest
length such thatσ(λ) = ν1.

Let λ be a dominant weight. Then we define a labeled directed graphΓ as follows. The points ofΓ are
the paths inΠλ. There is an edge with labelα ∈ ∆ from π1 to π2 if fαπ1 = π2. Now by [Kas96] this
graphΓ is isomorphic to the crystal graph of the highest-weight module with highest weightλ. So the
path model provides an efficient way of computing the crystal graph of a highest-weight module, with-
out constructing the module first. Also we see thatfαi1

· · · fαir
ξλ = 0 is equivalent tõFαi1

· · · F̃αir
vλ = 0,

wherevλ ∈V(λ) is a highest weight vector (or rather the image of it inL(λ)/qL(λ)), and theF̃αk are
the Kashiwara operators onB(λ) (see Section2.6).

2.8 Notes

I refer to [Hum90] for more information on Weyl groups, and to [Ste01] for an overview of algorithms
for computing with weights, Weyl groups and their elements.

For general introductions into the theory of quantized enveloping algebras I refer to [Car98],
[Jan96] (from where most of the material of this chapter is taken), [Lus92], [Lus93], [Ros91]. I
refer to the papers by Littelmann ([Lit94], [Lit95], [Lit98]) for more information on the path model.
The paper by Kashiwara ([Kas96]) contains a proof of the connection between path operators and
Kashiwara operators.

Finally, I refer to [Gra01] (on computing with PBW-type bases), [Gra02] (computation of elements
of the canonical basis) for an account of some of the algorithms used inQuaGroup.



Chapter 3

QuaGroup

In this chapter we describe the functionality provided byQuaGroup.

3.1 Global constants

3.1.1 QuantumField

♦ QuantumField (global variable)

This is the fieldQ(q) of rational functions inq, overQ.
Example

gap> QuantumField;
QuantumField

3.1.2 q

♦ q (global variable)

This is an indeterminate;QuantumField is the field of rational functions in this indeterminate.
The identifier q is fixed once the packageQuaGroup is loaded. The symbolq is chosen (instead of
q) in order to avoid potential name clashes. We note thatq is printed asq.

Example
gap> _q;
q
gap> _q in QuantumField;
true

3.2 Gaussian integers

3.2.1 GaussNumber

♦ GaussNumber( n, par ) (operation)

This function computes for the integern the Gaussian integer[n]v=par (cf. Section2.1).

14

Example
gap> GaussNumber( 4, _q );
qˆ-3+qˆ-1+q+qˆ3

3.2.2 GaussianFactorial

♦ GaussianFactorial( n, par ) (operation)

This function computes for the integern the Gaussian factorial[n]!v=par.
Example

gap> GaussianFactorial( 3, _q );
qˆ-3+2*qˆ-1+2*q+qˆ3
gap> GaussianFactorial( 3, _qˆ2 );
qˆ-6+2*qˆ-2+2*qˆ2+qˆ6

3.2.3 GaussianBinomial

♦ GaussianBinomial( n, k, par ) (operation)

This function computes for two integersn andk the Gaussian binomialn choosek, where the
parameterv is replaced bypar.

Example
gap> GaussianBinomial( 5, 2, _qˆ2 );
qˆ-12+qˆ-8+2*qˆ-4+2+2*qˆ4+qˆ8+qˆ12

3.3 Roots and root systems

In this section we describe some functions for dealing with root systems. These functions supplement
the ones already present in theGAP library.

3.3.1 RootSystem

♦ RootSystem( type, rank ) (operation)

♦ RootSystem( list ) (operation)

Heretype is a capital letter between"A" and"G", andrank is a positive integer (≥ 1 if type="A",
≥ 2 if type="B", "C",≥ 4 if type="D", 6,7,8 if type="E", 4 if type="F", and 2 iftype="G"). This
function returns the root system of typetype and rankrank. In the second formlist is a list of types
and ranks, e.g.,[ "B", 2, "F", 4, "D", 7 ].

The root system constructed by this function comes with he attributesPositiveRoots,
NegativeRoots, SimpleSystem, CartanMatrix, BilinearFormMat. Here the attribute
SimpleSystem contains a set of simple roots, written as unit vectors.PositiveRoots is a list of the
positive roots, written as linear combinations of the simple roots, and likewise forNegativeRoots.
CartanMatrix( R ) is the Cartan matrix of the root systemR, where the entry on position(i, j)
is given by〈αi ,α∨

j 〉 whereαi is the i-th simple root.BilinearFormMat( R ) is the matrix of the
bilinear form, where the entry on position(i, j) is given by(αi ,α j) (see Section2.2).

WeylGroup( R ) returns the Weyl group of the root systemR. We refer to theGAP reference
manual for an overview of the functions for Weyl groups in theGAP library. We mention the functions
ConjugateDominantWeight( W, wt ) (returns the dominant weight in theW-orbit of the weight
wt), andWeylOrbitIterator( W, wt ) (returns an iterator for theW-orbit containing the weight
wt). We write weights as integral linear combinations of fundamental weights, so inGAP weights are
represented by lists of integers (of length equal to the rank of the root system).

Also we mention the functionPositiveRootsAsWeights( R ) that returns the positive roots of
R written as weights, i.e., as linear combinations of the fundamental weights.

Example
gap> R:=RootSystem( [ "B", 2, "F", 4, "E", 6 ] );
<root system of type B2 F4 E6>
gap> R:= RootSystem( "A", 2 );
<root system of type A2>
gap> PositiveRoots( R );
[ [ 1, 0 ], [ 0, 1 ], [ 1, 1 ] ]
gap> BilinearFormMat( R );
[ [ 2, -1 ], [ -1, 2 ] ]
gap> W:= WeylGroup( R );
Group([ [ [ -1, 1 ], [ 0, 1 ] ], [ [ 1, 0 ], [ 1, -1 ] ] ])
gap> ConjugateDominantWeight( W, [-3,2] );
[ 2, 1 ]
gap> o:= WeylOrbitIterator( W, [-3,2] );
<iterator>
# Using the iterator we can loop over the orbit:
gap> NextIterator( o );
[ 2, 1 ]
gap> NextIterator( o );
[ -1, -2 ]
gap> PositiveRootsAsWeights( R );
[ [ 2, -1 ], [ -1, 2 ], [ 1, 1 ] ]

3.3.2 BilinearFormMatNF

♦ BilinearFormMatNF( R ) (attribute)

This is the matrix of the “normalized” bilinear form. This means that all diagonal entries are
even, and 2 is the minimum value occurring on the diagonal. IfR is a root system constructed by
RootSystem (3.3.1), then this is equal toBilinearFormMat( R ).

3.3.3 PositiveRootsNF

♦ PositiveRootsNF( R ) (attribute)

This is the list of positive roots of the root systemR, written as linear combinations of the sim-
ple roots. This means that the simple roots are unit vectors. IfR is a root system constructed by
RootSystem (3.3.1), then this is equal toPositiveRoots( R ).

One of the reasons for writing the positive roots like this is the following. Leta, b be two
elements ofPositiveRootsNF( R ), and letB be the matrix of the bilinear form. Thena*( B*b )
is the result of applying the bilinear form toa, b.

Example
gap> R:= RootSystem( SimpleLieAlgebra( "B", 2, Rationals ) );;
gap> PositiveRootsNF( R );
[ [ 1, 0 ], [ 0, 1 ], [ 1, 1 ], [ 1, 2 ] ]
# We note that in this case PositiveRoots( R ) will give the positive roots in
# a different format.

3.3.4 SimpleSystemNF

♦ SimpleSystemNF( R ) (attribute)

This is the list of simple roots ofR, written as unit vectors (this means that they are elements of
PositiveRootsNF( R )). If R is a root system constructed byRootSystem (3.3.1), then this is equal
to SimpleSystem( R ).

3.3.5 PositiveRootsInConvexOrder

♦ PositiveRootsInConvexOrder( R ) (attribute)

This function returns the positive roots of the root systemR, in the “convex” order. Letw0 = s1 · · ·st

be a reduced expression of the longest element in the Weyl group. Then thek-th element of the list
returned by this function iss1 · · ·sk−1(αk). (Where the reduced expression used is the one returned by
LongestWeylWord( R ).) If α, β andα + β are positive roots, thenα + β occurs betweenα andβ
(whence the name convex order).

In the output all roots are written in “normal form”, i.e., as elements ofPositiveRootsNF( R ).
Example

gap> R:= RootSystem( "G", 2 );;
gap> PositiveRootsInConvexOrder( R );
[ [ 1, 0 ], [ 3, 1 ], [ 2, 1 ], [ 3, 2 ], [ 1, 1 ], [ 0, 1 ] ]

3.3.6 SimpleRootsAsWeights

♦ SimpleRootsAsWeights( R ) (attribute)

Returns the simple roots of the root systemR, written as linear combinations of the fundamental
weights.

Example
gap> R:= RootSystem( "A", 2 );;
gap> SimpleRootsAsWeights( R );
[ [ 2, -1 ], [ -1, 2 ] ]

3.4 Weyl groups and their elements

Now we describe a few functions that deal with reduced words in the Weyl group of the root system
R. These words are represented as lists of positive integersi, denoting thei-th simple reflection
(which corresponds to thei-th element ofSimpleSystem( R )). For example[ 3, 2, 1, 3, 1 ]
represents the expressions3s2s1s3s1.

3.4.1 ApplyWeylElement

♦ ApplyWeylElement( W, wt, wd ) (operation)

Herewd is a (not necessarily reduced) word in the Weyl groupW, andwt is a weight (written as
integral linear combination of the simple weights). This function returns the result of applyingwd
to wt. For example, ifwt=µ, andwd = [ 1, 2 ] then this function returnss1s2(µ) (wheresi is the
simple reflection corresponding to thei-th simple root).

Example
gap> W:= WeylGroup( RootSystem( "G", 2 ) ) ;;
gap> ApplyWeylElement( W, [ -3, 7 ], [ 1, 1, 2, 1, 2 ] );
[ 15, -11 ]

3.4.2 LengthOfWeylWord

♦ LengthOfWeylWord( W, wd ) (operation)

Herewd is a word in the Weyl groupW. This function returns the length of that word.
Example

gap> W:= WeylGroup( RootSystem( "F", 4 ) ) ;
<matrix group with 4 generators>
gap> LengthOfWeylWord( W, [ 1, 3, 2, 4, 2 ] );
3

3.4.3 LongestWeylWord

♦ LongestWeylWord( R ) (attribute)

HereR is a root system.LongestWeylWord( R ) returns the longest word in the Weyl group of
R.

If this function is called for a root systemR, a reduced expression for the longest element in
the Weyl group is calculated (the one which is the smallest in the lexicographical ordering). How-
ever, if you would like to work with a different reduced expression, then it is possible to set it by
SetLongestWeylWord( R, wd ), wherewd is a reduced expression of the longest element in the
Weyl group. Note that you will have to do this before callingLongestWeylWord, or any function that
may callLongestWeylWord (once the attribute is set, it will not be possible to change it). Note also
that you must be sure that the word you give is in fact a reduced expression for the longest element in
the Weyl group, as this is not checked (you can check this withLengthOfWeylWord (3.4.2)).

We note that virtually all algorithms for quantized enveloping algebras depend on the choice of
reduced expression for the longest element in the Weyl group (as the PBW-type basis depends on
this).

Example
gap> R:= RootSystem( "G", 2 );;
gap> LongestWeylWord( R );
[ 1, 2, 1, 2, 1, 2 ]

3.4.4 ReducedWordIterator

♦ ReducedWordIterator( W, wd ) (operation)

HereW is a Weyl group, andwd a reduced word. This function returns an iterator for the set of
reduced words that represent the same element aswd. The elements are output in ascending lexico-
graphical order.

Example
gap> R:= RootSystem( "F", 4 );;
gap> it:= ReducedWordIterator( WeylGroup(R), LongestWeylWord(R) );
<iterator>
gap> NextIterator( it );
[ 1, 2, 1, 3, 2, 1, 3, 2, 3, 4, 3, 2, 1, 3, 2, 3, 4, 3, 2, 1, 3, 2, 3, 4 ]
gap> k:= 1;;
gap> while not IsDoneIterator( it ) do
> k:= k+1; w:= NextIterator( it );
> od;
gap> k;
2144892

So there are 2144892 reduced expressions for the longest element in the Weyl group of typeF4.

3.4.5 ExchangeElement

♦ ExchangeElement( W, wd, ind ) (operation)

HereW is a Weyl group, andwd is areducedword inW, andind is an index between 1 and the rank
of the root system. Letv denote the word obtained fromwd by addingind at the end. This function
assumesthat the length ofv is one less than the length ofwd, and returns a reduced expression forv
that is obtained fromwd by deleting one entry. Nothing is guaranteed of the output if the length ofv
is bigger than the length ofwd.

Example
gap> R:= RootSystem( "G", 2 );;
gap> wd:= LongestWeylWord( R );;
gap> ExchangeElement( WeylGroup(R), wd, 1 );
[ 2, 1, 2, 1, 2 ]

3.4.6 GetBraidRelations

♦ GetBraidRelations( W, wd1, wd2 ) (operation)

HereW is a Weyl group, andwd1, wd2 are two reduced words representing the same element in
W. This function returns a list of braid relations that can be applied towd1 to obtainwd2. Here a
braid relation is represented as a list, with at the odd positions integers that represent positions in a
word, and at the even positions the indices that are on those positions after applying the relation. For
example, letwd be the word[ 1, 2, 1, 3, 2, 1 ] and letr = [ 3, 3, 4, 1 ] be a relation.
Then the result of applyingr to wd is [ 1, 2, 3, 1, 2, 1] (i.e., on the third position we put a 3,
and on the fourth position a 1).

We note that the function does not check first whetherwd1 andwd2 represent the same element
in W. If this is not the case, then an error will occur during the execution of the function, or it will
produce wrong output.

Example
gap> R:= RootSystem( "A", 3 );;
gap> wd1:= LongestWeylWord( R );
[ 1, 2, 1, 3, 2, 1 ]
gap> wd2:= [ 1, 3, 2, 1, 3, 2 ];;
gap> GetBraidRelations( WeylGroup(R), wd1, wd2 );
[ [ 3, 3, 4, 1 ], [ 4, 2, 5, 1, 6, 2 ], [ 2, 3, 3, 2, 4, 3 ],

[ 4, 1, 5, 3 ] ]

3.4.7 LongWords

♦ LongWords( R ) (attribute)

For a root systemR this returns a list of triples (of length equal to the rank ofR). Let t be thek-th
triple occurring in this list. The first element oft is an expression for the longest element of the Weyl
group, starting withk. The second element is a list of braid relations, moving this expression to the
value ofLongestWeylWord( R ). The third element is a list of braid relations performing the reverse
transformation.

Example
gap> R:= RootSystem( "A", 3 );;
gap> LongWords( R )[3];
[ [ 3, 1, 2, 1, 3, 2 ],

[ [ 3, 3, 4, 1 ], [ 4, 2, 5, 1, 6, 2 ], [ 2, 3, 3, 2, 4, 3 ],
[ 4, 1, 5, 3 ], [ 1, 3, 2, 1 ] ],

[ [ 4, 3, 5, 1 ], [ 1, 1, 2, 3 ], [ 2, 2, 3, 3, 4, 2 ],
[ 4, 1, 5, 2, 6, 1 ], [ 3, 1, 4, 3 ] ] ]

3.5 Quantized enveloping algebras

In QuaGroup we deal with two types of quantized enveloping algebra. First there are the quantized
enveloping algebras defined over the fieldQuantumField (3.1.1). We say that these algebras are
“generic” quantized enveloping algebras, inQuaGroup they have the categoryIsGenericQUEA. Sec-
ondly, we deal with the quantized enveloping algebras that are defined over a different field.

3.5.1 QuantizedUEA

♦ QuantizedUEA( R ) (attribute)

♦ QuantizedUEA( R, F, v ) (operation)

♦ QuantizedUEA( L ) (attribute)

♦ QuantizedUEA( L, F, v ) (operation)

In the first two formsR is a root system. With onlyR as input, the corresponding generic quantized
enveloping algebra is constructed. It is stored as an attribute ofR (so that constructing it twice for the
same root system yields the same object). Also the root system is stored in the quantized enveloping
algebra as the attributeRootSystem.

The attributeGeneratorsOfAlgebra contains the generators of a PBW-type basis (see Section
2.4), that are constructed relative to the reduced expression for the longest element in the Weyl group
that is contained inLongestWeylWord( R ). We refer toObjByExtRep (3.5.2) for a description of
the construction of elements of a quantized enveloping algebra.

The call QuantizedUEA( R, F, v ) returns the quantized universal enveloping algebra
with quantum parameterv, which must lie in the fieldF. In this case the elements of
GeneratorsOfAlgebra are the images of the generators of the corresponding generic quantized en-
veloping algebra. This means that ifv is a root of unity, then the generators will not generate the whole
algebra, but rather a finite dimensional subalgebra (as for instanceEk

i = 0 for k large enough). It is
possible to construct elements that do not lie in this finite dimensional subalgebra usingObjByExtRep
(3.5.2).

In the last two casesL must be a semisimple Lie algebra. The two calls are short for
QuantizedUEA( RootSystem( L ) ) and QuantizedUEA( RootSystem( L ), F, v ) respec-
tively.

Example
# We construct the generic quantized enveloping algebra corresponding
# to the root system of type A2+G2:
gap> R:= RootSystem( [ "A", 2, "G", 2 ] );;
gap> U:= QuantizedUEA( R );
QuantumUEA( <root system of type A2 G2>, Qpar = q )
gap> RootSystem( U );
<root system of type A2 G2>
gap> g:= GeneratorsOfAlgebra( U );
[ F1, F2, F3, F4, F5, F6, F7, F8, F9, K1, K1+(qˆ-1-q)*[ K1 ; 1 ], K2,

K2+(qˆ-1-q)*[ K2 ; 1 ], K3, K3+(qˆ-1-q)*[ K3 ; 1 ], K4,
K4+(qˆ-3-qˆ3)*[ K4 ; 1 ], E1, E2, E3, E4, E5, E6, E7, E8, E9 ]

# These elements generate a PBW-type basis of U; the nine elements Fi,
# and the nine elements Ei correspond to the roots listed in convex order:
gap> PositiveRootsInConvexOrder( R );
[ [ 1, 0, 0, 0 ], [ 1, 1, 0, 0 ], [ 0, 1, 0, 0 ], [ 0, 0, 1, 0 ],

[ 0, 0, 3, 1 ], [ 0, 0, 2, 1 ], [ 0, 0, 3, 2 ], [ 0, 0, 1, 1 ],
[ 0, 0, 0, 1 ] ]

# So, for example, F5 is an element of weight -[ 0, 0, 3, 1 ].
# We can also multiply elements; the result is written on the PBW-basis:
gap> g[17]*g[4];
(qˆ-6-1)*F4*[ K4 ; 1 ]+(qˆ-3)*F4*K4
# Now we construct a non-generic quantized enveloping algebra:
gap> R:= RootSystem( "A", 2 );;
gap> U:= QuantizedUEA( R, CF(3), E(3) );;
gap> g:= GeneratorsOfAlgebra( U );
[ F1, F2, F3, K1, K1+(-E(3)+E(3)ˆ2)*[ K1 ; 1 ], K2,

K2+(-E(3)+E(3)ˆ2)*[ K2 ; 1 ], E1, E2, E3 ]

As can be seen in the example, every element ofU is written as a linear combination of monomials
in the PBW-generators; the generators ofU− come first, then the generators ofU0, and finally the
generators ofU+.

3.5.2 ObjByExtRep

♦ ObjByExtRep( fam, list ) (operation)

Herefam is the elements family of a quantized enveloping algebraU. Secondly,list is a list
describing an element ofU. We explain how this description works. First we describe an indexing
system for the generators ofU. LetR be the root system ofU. Lett be the number of positive roots, and
rank the rank of the root system. Then the generators ofU areFk, Ki (and its inverse),Ek, for k=1...t,
i=1..rank. (See Section2.4; for the construction of theFk, Ek, the value ofLongestWeylWord( R )
is used.) Now the index ofFk is k, and the index ofEk is t+rank+k. Furthermore, elements of
the algebra generated by theKi, and its inverse, are written as linear combinations of products of
“binomials”, as in Section2.5. The element

Kd
i

[
Ki

s

]
(whered = 0,1), is indexed as[ t+i, d ] (what happens to thes is described later). So an index is
either an integer, or a list of two integers.

A monomial is a list of indices, each followed by an exponent. First come the indices of theFk,
(1..t), then come the lists of the form[ t+i, d ], and finally the indices of theEk. Each index
is followed by an exponent. An index of the form[ t+i, d ] is followed by thes in the above
formula.

The second argument ofObjByExtRep is a list of monomials followed by coefficients. This func-
tion returns the element ofU described by this list.

Finally we remark that the element

Kd
i

[
Ki

s

]
is printed asKi[ Ki ; s ] if d=1, and as[ Ki ; s ] if d=0.

Example
gap> U:= QuantizedUEA( RootSystem("A",2) );;
gap> fam:= ElementsFamily( FamilyObj( U ) );;
gap> list:= [ [ 2, 3, [ 4, 0 ], 8, 6, 11 ], _qˆ2, # monomial and coefficient
> [ 1, 7, 3, 5, [ 5, 1 ], 3, 8, 9 ], _qˆ-1 + _qˆ2 ]; # monomial and coefficient
[ [ 2, 3, [ 4, 0 ], 8, 6, 11 ], qˆ2, [ 1, 7, 3, 5, [ 5, 1 ], 3, 8, 9 ],

qˆ-1+qˆ2 ]
gap> ObjByExtRep( fam, list );
(qˆ2)*F2ˆ(3)*[ K1 ; 8 ]*E1ˆ(11)+(qˆ-1+qˆ2)*F1ˆ(7)*F3ˆ(5)*K2[ K2 ; 3 ]*E3ˆ(9)

3.5.3 ExtRepOfObj

♦ ExtRepOfObj( elm ) (operation)

For the elementelm of a quantized enveloping algebra, this function returns the list that defines
elm (seeObjByExtRep (3.5.2)).

Example
gap> U:= QuantizedUEA( RootSystem("A",2) );;
gap> g:= GeneratorsOfAlgebra(U);
[ F1, F2, F3, K1, K1+(qˆ-1-q)*[ K1 ; 1 ], K2, K2+(qˆ-1-q)*[ K2 ; 1 ], E1,

E2, E3 ]
gap> ExtRepOfObj( g[5] );
[ [ [ 4, 1 ], 0 ], 1, [ [ 4, 0 ], 1 ], qˆ-1-q ]

3.5.4 QuantumParameter

♦ QuantumParameter( U ) (attribute)

Returns the quantum parameter used in the definition ofU.
Example

gap> R:= RootSystem("A",2);;
gap> U0:= QuantizedUEA( R, CF(3), E(3) );;
gap> QuantumParameter( U0 );
E(3)

3.5.5 CanonicalMapping

♦ CanonicalMapping( U ) (attribute)

HereU is a quantized enveloping algebra. LetU0 denote the corresponding “generic” quantized
enveloping algebra. This function returns the mappingU0 --> U obtained by mappingq (which is
the quantum parameter ofU0) to the quantum parameter ofU.

Example
gap> R:= RootSystem("A", 3 );;
gap> U:= QuantizedUEA( R, CF(5), E(5) );;
gap> f:= CanonicalMapping( U );
MappingByFunction( QuantumUEA( <root system of type A
3>, Qpar = q ), QuantumUEA( <root system of type A3>, Qpar =
E(5) ), function( u ) ... end )
gap> U0:= Source( f );
QuantumUEA( <root system of type A3>, Qpar = q )
gap> g:= GeneratorsOfAlgebra( U0 );;
gap> u:= g[18]*g[9]*g[6];
(qˆ2)*F6*K2*E6+(q)*K2*[ K3 ; 1 ]
gap> Image( f, u );
(E(5)ˆ2)*F6*K2*E6+(E(5))*K2*[ K3 ; 1 ]

3.5.6 WriteQEAToFile

♦ WriteQEAToFile( U, file ) (operation)

HereU is a quantized enveloping algebra, and file is a string containing the name of a file. This
function writes some data tofile, that allowsReadQEAFromFile (3.5.7) to recover it.

Example
gap> U:= QuantizedUEA( RootSystem("A",3) );;
gap> WriteQEAToFile( U, "/home/wdg/A3" );

3.5.7 ReadQEAFromFile

♦ ReadQEAFromFile( file ) (operation)

Herefile is a string containing the name of a file, to which a quantized enveloping algebra has
been written byWriteQEAToFile (3.5.6). This function recovers the quantized enveloping algebra.

Example
gap> U:= QuantizedUEA( RootSystem("A",3) );;
gap> WriteQEAToFile( U, "/home/wdg/A3" );
gap> U0:= ReadQEAFromFile( "/home/wdg/A3" );
QuantumUEA( <root system of type A3>, Qpar = q )

3.6 Homomorphisms and automorphisms

Here we describe functions for creating homomorphisms and (anti)-automorphisms of a quantized
enveloping algebra.

3.6.1 QEAHomomorphism

♦ QEAHomomorphism( U, A, list ) (operation)

HereU is a generic quantized enveloping algebra (i.e., with quantum parameterq), A is an algebra
with one overQuantumField, andlist is a list of 4*rank elements ofA (whererank is the rank
of the root system ofU). On the first rank positions there are the images of theFα (where theα
are simple roots, listed in the order in which they occur inSimpleSystem( R )). On the positions
rank+1...2*rank are the images of theKα. On the positions2*rank+1...3*rank are the images
of theK−1

α , and finally on the positions3*rank+1...4*rank occur the images of theEα.
This function returns the homomorphismU -> A, defined by this data. In the example below we

construct a homomorphism from one quantized enveloping algebra into another. Both are constructed
relative to the same root system, but with different reduced expressions for the longest element of the
Weyl group.

Example
gap> R:= RootSystem( "G", 2 );;
gap> SetLongestWeylWord( R, [1,2,1,2,1,2] );
gap> UR:= QuantizedUEA( R );;
gap> S:= RootSystem( "G", 2 );;
gap> SetLongestWeylWord( S, [2,1,2,1,2,1] );
gap> US:= QuantizedUEA( S );;
gap> gS:= GeneratorsOfAlgebra( US );
[ F1, F2, F3, F4, F5, F6, K1, K1+(qˆ-1-q)*[ K1 ; 1 ], K2,

K2+(qˆ-3-qˆ3)*[ K2 ; 1 ], E1, E2, E3, E4, E5, E6 ]
gap> SimpleSystem( R );
[ [ 1, 0 ], [ 0, 1 ] ]
gap> PositiveRootsInConvexOrder( S );
[ [ 0, 1 ], [ 1, 1 ], [ 3, 2 ], [ 2, 1 ], [ 3, 1 ], [ 1, 0 ] ]
# We see that the simple roots of R occur on positions 6 and 1
# in the list PositiveRootsInConvexOrder( S ); This means that we
# get the following list of images of the homomorphism:
gap> imgs:= [ gS[6], gS[1], # the images of the F_{\alpha}
> gS[7], gS[9], # the images of the K_{\alpha}
> gS[8], gS[10], # the images of the K_{\alpha}ˆ{-1}
> gS[16], gS[11] ]; # the images of the E_{\alpha}
[ F6, F1, K1, K2, K1+(qˆ-1-q)*[ K1 ; 1 ], K2+(qˆ-3-qˆ3)*[ K2 ; 1 ], E6, E1
]
gap> h:= QEAHomomorphism( UR, US, imgs );
<homomorphism: QuantumUEA( <root system of type G

2>, Qpar = q ) -> QuantumUEA( <root system of type G2>, Qpar = q )>
gap> Image( h, GeneratorsOfAlgebra( UR )[3] );
(1-qˆ4-qˆ6+qˆ10)*F1*F6ˆ(2)+(-qˆ2+qˆ6)*F2*F6+(qˆ4)*F4

3.6.2 QEAAutomorphism

♦ QEAAutomorphism( U, list ) (operation)

♦ QEAAutomorphism( U, f ) (operation)

In the first formU is a generic quantized enveloping algebra (i.e., with quantum parameterq), and
list is a list of4*rank elements ofU (whererank is the rank of the corresponding root system). On
the firstrank positions there are the images of theFα (where theα are simple roots, listed in the order
in which they occur inSimpleSystem( R )). On the positionsrank+1...2*rank are the images of
theKα. On the positions2*rank+1...3*rank are the images of theK−1

α , and finally on the positions
3*rank+1...4*rank occur the images of theEα.

In the second formU is a non-generic quantized enveloping algebra, andf is an automorphism
of the corresponding generic quantized enveloping algebra. The corresponding automorphism ofU is
constructed. In this casef must not be the bar-automorphism of the corresponding generic quantized
enveloping algebra (cf.BarAutomorphism (3.6.6)), as this automorphism doesn’t work in the non-
generic case.

The image of an elementx under an automorphismf is computed byImage( f, x ). Note
that there is no function for calculating pre-images (in general this seems to be a very hard
task). If you want the inverse of an automorphism, you have to construct it explicitly (e.g., by
QEAAutomorphism( U, list ), wherelist is a list of pre-images).

Below we construct the automorphismω (cf. Section2.2) of the quantized enveloping of typeA3,
when the quantum parameter isq, and when the quantum parameter is a fifth root of unity.

Example
# First we construct the quantized enveloping algebra:
gap> R:= RootSystem( "A", 3 );;
gap> U0:= QuantizedUEA( R );
QuantumUEA( <root system of type A3>, Qpar = q )
gap> g:= GeneratorsOfAlgebra( U0 );
[ F1, F2, F3, F4, F5, F6, K1, K1+(qˆ-1-q)*[ K1 ; 1 ], K2,

K2+(qˆ-1-q)*[ K2 ; 1 ], K3, K3+(qˆ-1-q)*[ K3 ; 1 ], E1, E2, E3, E4, E5, E6 ]
# Now, for instance, we map F_{\alpha} to E_{\alpha}, where \alpha
# is a simple root. In order to find where those F_{\alpha}, E_{\alpha}
# are in the list of generators, we look at the list of positive roots
# in convex order:
gap> PositiveRootsInConvexOrder( R );
[ [ 1, 0, 0 ], [ 1, 1, 0 ], [ 0, 1, 0 ], [ 1, 1, 1 ], [ 0, 1, 1 ],

[ 0, 0, 1 ] ]
# So the simple roots occur on positions 1, 3, 6. This means that we
# have the following list of images:
gap> imgs:= [ g[13], g[15], g[18], g[8], g[10], g[12], g[7], g[9], g[11],
> g[1], g[3], g[6] ];
[ E1, E3, E6, K1+(qˆ-1-q)*[ K1 ; 1 ], K2+(qˆ-1-q)*[ K2 ; 1 ],

K3+(qˆ-1-q)*[ K3 ; 1 ], K1, K2, K3, F1, F3, F6 ]
gap> f:= QEAAutomorphism( U0, imgs );
<automorphism of QuantumUEA( <root system of type A3>, Qpar = q )>
gap> Image( f, g[2] );

(-q)*E2
# f induces an automorphism of any non-generic quantized enveloping
# algebra with the same root system R:
gap> U1:= QuantizedUEA( R, CF(5), E(5) );
QuantumUEA( <root system of type A3>, Qpar = E(5) )
gap> h:= QEAAutomorphism( U1, f );
<automorphism of QuantumUEA( <root system of type A3>, Qpar = E(5) )>
gap> Image( h, GeneratorsOfAlgebra(U1)[7] );
(-E(5)+E(5)ˆ4)*[ K1 ; 1 ]+K1

3.6.3 QEAAntiAutomorphism

♦ QEAAntiAutomorphism( U, list ) (operation)

♦ QEAAntiAutomorphism( U, f ) (operation)

These are functions for constructing anti-automorphisms of quantized enveloping algebras. The
same comments apply as forQEAAutomorphism (3.6.2).

3.6.4 AutomorphismOmega

♦ AutomorphismOmega( U ) (attribute)

This is the automorphismω (cf. Section2.2).
Example

gap> R:= RootSystem( "A", 3 );;
gap> U:= QuantizedUEA( R, CF(5), E(5) );
QuantumUEA( <root system of type A3>, Qpar = E(5) )
gap> f:= AutomorphismOmega( U );
<automorphism of QuantumUEA( <root system of type A3>, Qpar = E(5) )>

3.6.5 AntiAutomorphismTau

♦ AntiAutomorphismTau( ) (attribute)

This is the anti-automorphismτ (cf. Section2.2).
Example

gap> R:= RootSystem( "A", 3 );;
gap> U:= QuantizedUEA( R, CF(5), E(5) );
QuantumUEA( <root system of type A3>, Qpar = E(5) )
gap> t:= AntiAutomorphismTau( U );
<anti-automorphism of QuantumUEA( <root system of type A3>, Qpar = E(5) )>

3.6.6 BarAutomorphism

♦ BarAutomorphism( U ) (attribute)

This is the automorphism ¯ defined in Section2.2HereU must be a generic quantized enveloping
algebra.

Example
gap> U:= QuantizedUEA( RootSystem(["A",2,"B",2]) );;
gap> bar:= BarAutomorphism( U );
<automorphism of QuantumUEA( <root system of type A2 B2>, Qpar = q )>
gap> Image( bar, GeneratorsOfAlgebra( U )[5] );
(-qˆ-2+qˆ2)*F4*F7+F5

3.6.7 AutomorphismTalpha

♦ AutomorphismTalpha( U, ind ) (operation)

This is the automorphismTα (cf. Section2.4), whereα is theind-th simple root.
Example

gap> U:= QuantizedUEA( RootSystem( "B", 3 ) );;
gap> f:=AutomorphismTalpha( U, 1 );
<automorphism of QuantumUEA( <root system of type B3>, Qpar = q )>
gap> a:= GeneratorsOfAlgebra( U )[3];
F3
gap> Image( f, a );
F2

3.6.8 DiagramAutomorphism

♦ DiagramAutomorphism( U, perm ) (operation)

This is the automorphism ofU induced by a diagram automorphism of the underlying root system.
The diagram automorphism is represented by the permutationperm, which is the permutation of the
simple roots performed by the diagram automorphism.

In the example below we construct the diagram automorphism of the root system of typeA3, which
is represented by the permutation(1,3).

Example
gap> R:= RootSystem( "A", 3 );;
gap> U:= QuantizedUEA( R );;
gap> f:= DiagramAutomorphism( U, (1,3) );
<automorphism of QuantumUEA( <root system of type A3>, Qpar = q )>
gap> g:= GeneratorsOfAlgebra( U );
[ F1, F2, F3, F4, F5, F6, K1, K1+(qˆ-1-q)*[ K1 ; 1 ], K2,

K2+(qˆ-1-q)*[ K2 ; 1 ], K3, K3+(qˆ-1-q)*[ K3 ; 1 ], E1, E2, E3, E4, E5, E6
]
gap> Image( f, g[1] );
F6

3.6.9 \ *

♦ \*( f, h ) (operation)

We can compose automorphisms and anti-automorphisms using the infix* operator. The result of
composing two automorphisms is an automorphism. The result of composing an automorphism and
an anti-automorphism is an anti-automorphism. The result of composing two anti-automorphisms is
an automorphism.

Example
gap> U:= QuantizedUEA( RootSystem( "B", 3 ) );;
gap> f:=AutomorphismTalpha( U, 1 );
<automorphism of QuantumUEA( <root system of type B3>, Qpar = q )>
gap> h:= AutomorphismOmega( U );
<automorphism of QuantumUEA( <root system of type B3>, Qpar = q )>
gap> f*h;
<automorphism of QuantumUEA( <root system of type B3>, Qpar = q )>
gap> t:= AntiAutomorphismTau( U );;
gap> T:= AutomorphismTalpha( U, 2 );;
gap> Tinv:= t*T*t;
<automorphism of QuantumUEA( <root system of type B3>, Qpar = q )>
# (The last call may take a little while.)
gap> x:= Image( T, GeneratorsOfAlgebra( U )[1] );
(1-qˆ4)*F1*F3+(-qˆ2)*F2
gap> Image( Tinv, x );
F1

According to [Jan96], 8.14(10),τ◦Tα ◦ τ is the inverse ofTα.

3.7 Hopf algebra structure

Here we describe functions for dealing with the Hopf algebra structure of a quantized enveloping
algebra. This structure enables us to construct tensor products, and dual modules of modules over
a quantized enveloping algebra. We refer to the next section (Section3.8) for some functions for
creating modules.

3.7.1 TensorPower

♦ TensorPower( U, d ) (operation)

HereU is a quantized universal enveloping algebra, andd a non-negative integer. This function
returns the associative algebra with underlying vector space thed-fold tensor product ofU with itself.
The product is defined component wise.

Example
gap> U:= QuantizedUEA( RootSystem( [ "B", 2 ] ) );;
gap> T:= TensorPower( U, 3 );
<algebra over QuantumField, with 36 generators>
gap> g:= GeneratorsOfAlgebra( T );;
gap> x:= g[1];
1*(1<x>1<x>F1)
gap> y:= g[30];
1*(E2<x>1<x>1)
gap> x*y;
1*(E2<x>1<x>F1)

3.7.2 UseTwistedHopfStructure

♦ UseTwistedHopfStructure( U, f, finv ) (operation)

HereU is a quantized enveloping algebra, andf, finv two (anti-) automorphisms ofU, wherefinv
is the inverse off. After calling this function the Hopf structure onU is used that is obtained from the
“normal” Hopf structure (see Section2.3) by twisting it withf.

A call to this function sets the attributeHopfStructureTwist, which is the list[ f, finv ].
Example

gap> U:= QuantizedUEA( RootSystem("A",2), CF(5), E(5) );;
gap> t:= AntiAutomorphismTau( U );;
gap> UseTwistedHopfStructure( U, t, t );

3.7.3 ComultiplicationMap

♦ ComultiplicationMap( U, d ) (operation)

This is a homomorphism from the quantized enveloping algebraU to thed-fold tensor power ofU
with itself. It is obtained by a repeated application of the comultiplication ofU. So ford=2 we get the
comultiplication ofU.

Example
gap> U:= QuantizedUEA( RootSystem("A",2), CF(5), E(5) );;
gap> D:= ComultiplicationMap( U, 3 );
<Comultiplication of QuantumUEA( <root system of type A2>, Qpar =
E(5) ), degree 3>
gap> Image( D, GeneratorsOfAlgebra(U)[4] );
1*(K1<x>K1<x>K1)

3.7.4 AntipodeMap

♦ AntipodeMap( U ) (attribute)

This is the antipode map of the quantized enveloping algebraU, which is constructed as an anti-
automorphism ofU.

Example
gap> U:= QuantizedUEA( RootSystem("A",2), CF(5), E(5) );;
gap> a:= AntipodeMap( U );
<anti-automorphism of QuantumUEA( <root system of type A2>, Qpar = E(5) )>

3.7.5 CounitMap

♦ CounitMap( U ) (attribute)

This is the counit map of the quantized enveloping algebraU, which is constructed as a function
from U to the ground field.

Example
gap> U:= QuantizedUEA( RootSystem("A",2), CF(5), E(5) );;
gap> co:= CounitMap( U );
function( u ) ... end
gap> x:= GeneratorsOfAlgebra( U )[4];
K1
gap> co( x );
1

3.8 Modules

Here we describe some functions for constructing left modules over quantized enveloping algebras.
We refer to theGAP reference manual for an overview of basic functions for algebra modules, which
are also applicable to the modules constructed by the functions described in this section. We mention
MatrixOfAction, DirectSumOfAlgebraModules. The action of an element of the algebra on an
element of the module is calculated by the infix operatorˆ.

3.8.1 HighestWeightModule (for a quantized env. alg.)

♦ HighestWeightModule( U, wt ) (operation)

HereU is a quantized universal enveloping algebra, andwt a dominant weight (i.e., a list of length
equal to the rank of the root system, consisting of non-negative integers). This function returns a
finite-dimensional highest-weight module of highest weightwt overU. If U is generic then this is the
unique irreducible highest-weight module overU. Otherwise it is the Weyl module, cf. Section2.5. In
this last case the module is not necessarily irreducible.

Let V denote the module returned by this function. The first basis element of the attribute
Basis( V ) is a highest-weight vector; it is written as1*v0. Other basis elements are written as, for
example,F2*F9*v0, which means that this vector is the result of letting the PBW-monomialF2*F9
act on the highest-weight vector.

Example
gap> U:= QuantizedUEA( RootSystem( [ "A", 2, "G", 2 ] ) );;
gap> V:= HighestWeightModule( U, [ 0, 1, 0, 2 ] );
<231-dimensional left-module over QuantumUEA( <root system of type A2 G
2>, Qpar = q )>
gap> Basis( V )[1];
1*v0
gap> Basis(V)[23]+(_qˆ2+_qˆ-2)*Basis(V)[137];
F3*F5*v0+(qˆ-2+qˆ2)*F8ˆ(6)*v0
# We compute the action of an element on a vector:
gap> gg:= GeneratorsOfAlgebra( U );;
gap> x:= gg[21]*gg[5];
F5*E4+(-qˆ-1)*F6*K3
gap> xˆBasis(V)[1];
(-qˆ-1)*F6*v0

3.8.2 IrreducibleQuotient

♦ IrreducibleQuotient( V ) (attribute)

HereV is a highest-weight module over a non-generic quantized enveloping algebra. This function
returns the quotient ofV by the maximal submodule not containing the highest weight vector. This is
not necessarily equal toV if the quantum parameter is a root of 1.

Example
gap> R:= RootSystem( "A", 2 );;
gap> U:= QuantizedUEA( R, CF(3), E(3) );;
gap> V:= HighestWeightModule( U, [1,1] );
<8-dimensional left-module over QuantumUEA( <root system of type A2>, Qpar =

E(3) )>
gap> IrreducibleQuotient( V );
<7-dimensional left-module over QuantumUEA( <root system of type A2>, Qpar =
E(3) )>

3.8.3 HWModuleByTensorProduct

♦ HWModuleByTensorProduct( U, wt ) (operation)

HereU must be agenericquantized enveloping algebra, andwt a dominant weight. This func-
tion returns the irreducible highest-weight module with highest weightwt. The algorithm uses
tensor products (whence the name). On some inputs this algorithm is faster than the one use for
HighestWeightModule:for a quantized env. alg. (3.8.1), on some inputs it is slower. I do
not know any good heuristics.

The basis supplied with the module returned is the canonical basis.
Example

gap> U:= QuantizedUEA( RootSystem("G",2) );;
gap> V:= HWModuleByTensorProduct( U, [2,1] );
<189-dimensional left-module over QuantumUEA( <root system of type G
2>, Qpar = q )>
# (This is a case where this algorithm is a lot faster.)

3.8.4 DIYModule

♦ DIYModule( U, V, acts ) (operation)

Here U is a generic quantized enveloping algebra, andV is a vector space over the field
QuantumField. U acts onV and the action is described by the data in the listacts. acts is a list
of lists, of length4*l, wherel is the rank of the root system.acts describes the actions of the gener-
ators[F1, ...,Fl ,K1, ...,Kl ,K

−
1 1, ...,K−

l 1,E1, ...,El ]. (HereFk is the generatorFαk, whereαk is thek-th
simple root, and likewise forEk.) The action of each generator is described by a list of lengthdim V,
giving the images of the basis elements ofV. If an image is zero then it may be omitted: in that case
there is a “hole” in the list. This function returns theU-module defined by the input.

LetR be a root system of typeA1, andU the corresponding quantized enveloping algebra (generated
by F,K,K−1,E). In the example below we construct the 2-dimensionalU-module with basis vectors
v1,v2, andU-action given byFv1 = v2, Fv2 = 0, Kv1 = qv1, Kv2 = q−1v2, Ev1 = 0, Ev2 = v1.

Example
gap> U:= QuantizedUEA( RootSystem("A",1) );
QuantumUEA( <root system of type A1>, Qpar = q )
gap> V:= QuantumFieldˆ2;
( QuantumFieldˆ2 )
gap> v:= BasisVectors( Basis(V) );
[ [ 1, 0 ], [ 0, 1 ] ]
gap> acts:= [ [ v[2], 0*v[1] ], [ _q*v[1], _qˆ-1*v[2] ],
> [ _qˆ-1*v[1], _q*v[2] ], [ 0*v[1], v[1] ] ];;
gap> M:= DIYModule( U, V, acts );
<2-dimensional left-module over QuantumUEA( <root system of type A
1>, Qpar = q )>

3.8.5 TensorProductOfAlgebraModules

♦ TensorProductOfAlgebraModules( V, W ) (operation)

♦ TensorProductOfAlgebraModules( V, W ) (operation)

Here V and W are two modules over the same quantized enveloping algebraU. This function
constructs the tensor product ofV andW (as aU-module). For this the comultiplication map ofU is
used (seeComultiplicationMap (3.7.3)).

In the second form list is a list ofU-modules. In that case the iterated tensor product is constructed.
Example

gap> U:= QuantizedUEA( RootSystem( [ "A", 2 ] ) );;
gap> V1:= HighestWeightModule( U, [ 1, 0 ] );;
gap> V2:= HighestWeightModule( U, [ 0, 1 ] );;
gap> TensorProductOfAlgebraModules( V1, V2 );
<9-dimensional left-module over QuantumUEA( <root system of type A2>, Qpar = q )>

3.8.6 HWModuleByGenerator

♦ HWModuleByGenerator( V, v, hw ) (operation)

HereV is a module over a generic quantized enveloping algebraU, v is a highest-weight vector (i.e.,
all Eαv=0), of weighthw, which must be dominant. This function returns a highest-weight module
overU isomorphic to the submodule ofV generated byv.

Example
gap> U:= QuantizedUEA( RootSystem("B",2) );;
gap> W1:= HighestWeightModule( U, [1,0] );;
gap> W2:= HighestWeightModule( U, [0,1] );;
gap> T:= TensorProductOfAlgebraModules( W1, W2 );
<20-dimensional left-module over QuantumUEA( <root system of type B
2>, Qpar = q )>
gap> HWModuleByGenerator( T, Basis(T)[1], [1,1] );
<16-dimensional left-module over QuantumUEA( <root system of type B
2>, Qpar = q )>

3.8.7 InducedQEAModule

♦ InducedQEAModule( U, V ) (operation)

HereU is a non-generic quantized enveloping algebra, andV a module over the corresponding
generic quantized enveloping algebra. This function returns theU-module obtained fromV by setting
q equal to the quantum parameter ofU.

Example
gap> R:= RootSystem("B",2);;
gap> U:= QuantizedUEA( R );;
gap> U0:= QuantizedUEA( R, CF(3), E(3) );;
gap> V:= HighestWeightModule( U, [1,1] );;
gap> W:= InducedQEAModule( U0, V );
<16-dimensional left-module over QuantumUEA( <root system of type B
2>, Qpar = E(3) )>

# This module is isomorphic to the one obtained by
# HighestWeightModule( U0, [1,1] );

3.8.8 GenericModule

♦ GenericModule( W ) (attribute)

For an induced module (seeInducedQEAModule (3.8.7)) this function returns the corresponding
module over the generic quantized enveloping algebra.

3.8.9 CanonicalMapping

♦ CanonicalMapping( W ) (attribute)

Here W is an induced module. LetV be the corresponding generic module (GenericModule
(3.8.8)). This function returns the mapV --> W, that sets q equal to the quantum parameter of
the acting algebra ofW.

Example
gap> R:= RootSystem("B",2);;
gap> U:= QuantizedUEA( R );;
gap> U0:= QuantizedUEA( R, CF(3), E(3) );;
gap> V:= HighestWeightModule( U, [1,1] );;
gap> W:= InducedQEAModule( U0, V );;
gap> f:= CanonicalMapping( W );
MappingByFunction( <
16-dimensional left-module over QuantumUEA( <root system of type B
2>, Qpar = q )>, <
16-dimensional left-module over QuantumUEA( <root system of type B
2>, Qpar = E(3) )>, function( v ) ... end )
gap> Image( f, _qˆ2*Basis(V)[3] );
(E(3)ˆ2)*e.3

3.8.10 U2Module

♦ U2Module( U, hw ) (operation)

HereU must be a quantized enveloping algebra of typeA2. This function returns the highest-
weight module overU of highest-weighthw (which must be dominant). This function is generally a
lot faster thanHighestWeightModule:for a quantized env. alg. (3.8.1).

Example
gap> U:= QuantizedUEA( RootSystem("A",2) );;
gap> A2Module( U, [4,7] );
<260-dimensional left-module over QuantumUEA( <root system of type A
2>, Qpar = q )>

3.8.11 MinusculeModule

♦ MinusculeModule( U, hw ) (operation)

HereU must be a generic quantized enveloping algebra, andhw a minuscule dominant weight. This
function returns the highest-weight module overU of highest-weighthw. This function is generally
somewhat faster thanHighestWeightModule:for a quantized env. alg. (3.8.1).

Example
gap> U:= QuantizedUEA( RootSystem("A",5) );;
gap> MinusculeModule( U, [0,0,1,0,0] );
<20-dimensional left-module over QuantumUEA( <root system of type A
5>, Qpar = q )>

3.8.12 DualAlgebraModule

♦ DualAlgebraModule( V ) (attribute)

Here V is a finite-dimensional left module over a quantized enveloping algebraU. This func-
tion returns the dual space ofV as an algebra module. For this the antipode map ofU is used (see
AntipodeMap (3.7.4)).

Let M denote the module returned by this function. ThenM has as basis the dual basis with respect
to Basis( V ). An element of this basis is printed asF@v, wherev is an element ofBasis( V ).
This is the function which takes the value1 onv and0 on all other basis elements. A general element
of M is a linear combination of these basis elements.

The elements ofM can be viewed as functions which take arguments. However, internally the
elements ofM are represented as wrapped up functions. The function corresponding to an elementm
of M is obtained byExtRepOfObj( m ) (the result of which is printed in the same way asm, but is not
equal to it).

Example
gap> U:= QuantizedUEA( RootSystem("A",2) );;
gap> V:= HighestWeightModule( U, [1,1] );;
gap> M:= DualAlgebraModule( V );
<8-dimensional left-module over QuantumUEA( <root system of type A
2>, Qpar = q )>
gap> u:= GeneratorsOfAlgebra( U )[2];
F2
gap> vv:= BasisVectors( Basis( M ) );
[ (1)*F@1*v0, (1)*F@F1*v0, (1)*F@F3*v0, (1)*F@F1*F3*v0, (1)*F@F2*v0,

(1)*F@F1*F2*v0, (1)*F@F2*F3*v0, (1)*F@F2ˆ(2)*v0 ]
gap> uˆvv[3];
<zero function>
# (The zero of the dual space is printed as <zero function>).
gap> uˆvv[4];
(qˆ3-qˆ5)*F@1*v0
# We get the function corresponding to a vector in M by using ExtRepOfObj:
gap> f:= ExtRepOfObj( vv[1] );
(1)*F@1*v0
# We can calculate images of this function:
gap> List( Basis(V), v -> Image( f, v ) );
[ 1, 0, 0, 0, 0, 0, 0, 0 ]

3.8.13 TrivialAlgebraModule

♦ TrivialAlgebraModule( U ) (attribute)

Returns the trivial module over the quantized enveloping algebraU. For this the counit map ofU
is used.

Example
gap> U:= QuantizedUEA( RootSystem("A",2) );;
gap> V:= TrivialAlgebraModule( U );
<left-module over QuantumUEA( <root system of type A2>, Qpar = q )>

3.8.14 WeightsAndVectors

♦ WeightsAndVectors( V ) (operation)

HereV is a left module over a quantized enveloping algebra.WeightsAndVectors( V ) is a list
of two lists; the first of these is a list of the weights ofV, the second a list of corresponding weight
vectors. These are again grouped in lists: if the multiplicity of a weight ism, then there arem weight
vectors, forming a basis of the corresponding weight space.

Modules constructed byHighestWeightModule:for a quantized env. alg. (3.8.1) come
with this attribute set. There is a method installed for computingWeightsAndVectors( V ), for
modulesV over a generic quantized enveloping algebra, such that all basis vectors (i.e., all elements
of Basis( V )) are weight vectors.

Example
gap> U:= QuantizedUEA( RootSystem( "A", 2 ) );;
gap> V:= HighestWeightModule( U, [ 1, 1 ] );;
gap> WeightsAndVectors( V );
[ [ [ 1, 1 ], [ -1, 2 ], [ 2, -1 ], [ 0, 0 ], [ -2, 1 ], [ 1, -2 ],

[ -1, -1 ] ],
[ [ 1*v0 ], [ F1*v0 ], [ F3*v0 ], [ F1*F3*v0, F2*v0 ], [ F1*F2*v0 ],

[ F2*F3*v0 ], [ F2ˆ(2)*v0 ] ] ]

3.8.15 HighestWeightsAndVectors

♦ HighestWeightsAndVectors( V ) (attribute)

Is analogous toWeightsAndVectors (3.8.14); now only the highest weights are listed along with
the corresponding highest-weight vectors.

There is a method installed for this usingWeightsAndVectors (3.8.14); which means that it
works if and only ifWeightsAndVectors( V ) works.

Example
gap> U:= QuantizedUEA( RootSystem( [ "A", 2 ] ) );;
gap> V:= HighestWeightModule( U, [ 1, 1 ] );;
gap> HighestWeightsAndVectors( V );
[ [ [ 1, 1 ] ], [ [ 1*v0 ] ] ]

3.8.16 RMatrix

♦ RMatrix( V ) (attribute)

HereV is a module over the a quantized enveloping algebraU. This function returns the matrix
of a linear mapθ : V⊗V →V⊗V that is a solution to the quantum Yang-Baxter equation. We have
that θ ◦P is an isomorphism ofU-modules, whereP : V ⊗V → V ⊗V is the linear map such that
P(v⊗w) = w⊗v. For more details we refer to [Jan96], Chapter 7.

This function works for modules for whichWeightsAndVectors (3.8.14) works.
Example

gap> U:= QuantizedUEA( RootSystem("A",1) );;
gap> V:= HighestWeightModule( U, [1] );;
gap> RMatrix( V );
[ [ 1, 0, 0, 0 ], [ 0, q, 1-qˆ2, 0 ], [ 0, 0, q, 0 ], [ 0, 0, 0, 1 ] ]

3.8.17 IsomorphismOfTensorModules

♦ IsomorphismOfTensorModules( V, W ) (operation)

HereV, W are two modules over the same quantized enveloping algebraU. This function returns a
linear mapθ : V⊗W →W⊗V that is an isomorphism of U-modules.

This function is only guaranteed to work correctly if the Hopf algebra structure is non-twisted (see
UseTwistedHopfStructure (3.7.2)).

This function works for modules for whichWeightsAndVectors (3.8.14) works.
Example

gap> U:= QuantizedUEA( RootSystem("B",2) );;
gap> V:= HighestWeightModule( U, [1,0] );;
gap> W:= HighestWeightModule( U, [0,1] );;
gap> h:= IsomorphismOfTensorModules( V, W );;
gap> VW:= Source( h );
<20-dimensional left-module over QuantumUEA( <root system of type B
2>, Qpar = q )>
gap> Image( h, Basis(VW)[13] );
q*(1*v0<x>F3*v0)+1-qˆ2*(F4*v0<x>F2*v0)+qˆ-1-qˆ3*(F3*v0<x>1*v0)

3.8.18 WriteModuleToFile

♦ WriteModuleToFile( V, file ) (operation)

HereV is a module over a quantized enveloping algebra, andfile is a string containing the name
of a file. This function writes some data tofile, that allowsReadModuleFromFile (3.8.19) to recover
it.

We remark that this function currently is only implemented for generic quantized enveloping
algebras.

3.8.19 ReadModuleFromFile

♦ ReadModuleFromFile( file ) (operation)

Herefile is a string containing the name of a file, to which a module over a quantized enveloping
algebra has been written byWriteModuleToFile (3.8.18). This function recovers the module. More
precisely: a new module is constructed that is isomorphic to the old one. In the process the algebra
acting on the module is constructed anew (from data written to the file). This algebra can be accessed
by LeftActingAlgebra( V ).

We remark that this function currently is only implemented for generic quantized enveloping
algebras.

Example
gap> U:= QuantizedUEA( RootSystem("A",3) );;
gap> V:= HighestWeightModule( U, [1,1,1] );;
gap> WriteModuleToFile( V, "/home/wdg/A3mod" );
gap> W:= ReadModuleFromFile( "/home/wdg/A3mod" );
<64-dimensional left-module over QuantumUEA( <root system of type A
3>, Qpar = q )>

3.9 The path model

In this section we describe functions for dealing with the path model. We work only with LS-paths,
which are represented by two lists, one of weights, and one of rationals (see Section2.7).

3.9.1 DominantLSPath

♦ DominantLSPath( R, wt ) (operation)

HereR is a root system, andwt a dominant weight in the weight lattice ofR. This function returns
the LS-path that is the line from the origin towt.

Example
gap> R:= RootSystem( "G", 2 );;
gap> DominantLSPath( R, [1,3] );
<LS path of shape [ 1, 3 ] ending in [ 1, 3 ] >

3.9.2 Falpha (for an LS-path)

♦ Falpha( path, ind ) (operation)

Is the result of applying the path operatorfαind
to the LS-pathpath (whereαind is theind-th

simple root).
The result isfail if fαind

(path)=0.

Example
gap> R:= RootSystem( "G", 2 );;
gap> p:=DominantLSPath( R, [1,3] );;
gap> p1:=Falpha( p, 1 );
<LS path of shape [ 1, 3 ] ending in [ -1, 4 ] >
gap> Falpha( p1, 1 );
fail

3.9.3 Ealpha (for an LS-path)

♦ Ealpha( path, ind ) (operation)

Is the result of applying the path operatoreαind
to the LS-pathpath (whereαind is theind-th

simple root).
The result isfail if eαind

(path)=0.

Example
gap> R:= RootSystem( "G", 2 );;
gap> p:=DominantLSPath( R, [1,3] );;
gap> Ealpha( p, 2 );
fail
gap> p1:=Falpha( p, 1 );;
gap> Ealpha( p1, 1 );
<LS path of shape [ 1, 3 ] ending in [ 1, 3 ] >

3.9.4 LSSequence

♦ LSSequence( path ) (attribute)

returns the two sequences (of weights and rational numbers) that define the LS-path path.
Example

gap> R:= RootSystem( "G", 2 );;
gap> p:=DominantLSPath( R, [1,3] );;
gap> p1:= Falpha( Falpha( p, 1 ), 2 );;
gap> LSSequence( p1 );
[ [ [ 11, -4 ], [ -1, 4 ] ], [ 0, 1/4, 1 ] ]

3.9.5 WeylWord

♦ WeylWord( path ) (attribute)

Herepath is an LS-path in the orbit (under the root operators) of a dominant LS-path ending in
the dominant weightλ. This means that the first direction of path is of the formw(λ) for somew in
the Weyl group. This function returns a list[i1, . . . , im] such thatw = si1 · · ·sim.

Example
gap> R:= RootSystem( "G", 2 );;
gap> p:=DominantLSPath( R, [1,3] );;
gap> p1:= Falpha( Falpha( Falpha( p, 1 ), 2 ), 1 );;
gap> WeylWord( p1 );
[ 1, 2, 1 ]

3.9.6 EndWeight

♦ EndWeight( path ) (attribute)

Herepath is an LS-path; this function returns the weight that is the endpoint of path

Example
gap> R:= RootSystem( "G", 2 );;
gap> p:=DominantLSPath( R, [1,3] );;
gap> p1:= Falpha( Falpha( Falpha( p, 1 ), 2 ), 1 );;
gap> EndWeight( p1 );
[ 0, 3 ]

3.9.7 CrystalGraph (for root system and weight)

♦ CrystalGraph( R, wt ) (function)

This function returns a record describing the crystal graph of the highest-weight module with
highest weightwt, over the quantized enveloping algebra corresponding toR. It is computed using the
path model. Therefore the points in the graph are LS-paths.

Denote the output byr; thenr.points is the list of points of the graph. Furthermore,r.edges is
a list of edges of the graph; this is a list of elements of the form[ [ i, j ], u ]. This means that
there is an arrow from pointi (i.e., the point on positioni in r.points) to pointj, with labelu.

Example
gap> R:= RootSystem( "A", 2 );;
gap> CrystalGraph( R, [1,1] );
rec(

points := [ <LS path of shape [ 1, 1 ] ending in [ 1, 1 ] >, <LS path of sha\
pe [ 1, 1 ] ending in [ -1, 2 ] >, <LS path of shape [ 1, 1 ] ending in

[ 2, -1 ] >, <LS path of shape [ 1, 1 ] ending in [ 0, 0 ] >,
<LS path of shape [ 1, 1 ] ending in [ 0, 0 ] >,
<LS path of shape [ 1, 1 ] ending in [ 1, -2 ] >,
<LS path of shape [ 1, 1 ] ending in [ -2, 1 ] >,
<LS path of shape [ 1, 1 ] ending in [ -1, -1 ] > ],

edges := [ [ [ 1, 2 ], 1 ], [ [ 1, 3 ], 2 ], [ [ 2, 4 ], 2 ],
[ [ 3, 5 ], 1 ], [ [ 4, 6 ], 2 ], [ [ 5, 7 ], 1 ], [ [ 6, 8 ], 1 ],
[ [ 7, 8 ], 2 ] ] )

3.10 Canonical bases

Here we describe functions for computing the canonical basis of the negative part of a quantized
enveloping algebra, and of a module.

3.10.1 Falpha (for a PBW-monomial)

♦ Falpha( x, ind ) (operation)

Herex is a PBW-monomial inU− (i.e., a monomial in theFα, whereα runs over the positive
roots). This function returns the result of applying theind-th Kashiwara operator̃Fαind

to x (cf.
Section2.6).

Example
gap> U:= QuantizedUEA( RootSystem( "F", 4 ) );;
gap> x:= One( U );
1

gap> Falpha( Falpha( x, 3 ), 2 );
F3*F9

3.10.2 Ealpha (for a PBW-monomial)

♦ Ealpha( x, ind ) (operation)

Herex is a PBW-monomial inU− (i.e., a monomial in theFα, whereα runs over the positive
roots). This function returns the result of applying theind-th Kashiwara operator̃Eαind

to x (cf.

Section2.6). The result isfail if Ẽαind
(x)=0.

Example
gap> U:= QuantizedUEA( RootSystem( "F", 4 ) );;
gap> Ealpha( One( U ), 2 );
fail
gap> g:= GeneratorsOfAlgebra( U );;
gap> x:= g[1]*g[4]*g[7]*g[17];
F1*F4*F7*F17
gap> Ealpha( x, 3 );
F1*F2*F7*F17

3.10.3 CanonicalBasis

♦ CanonicalBasis( U ) (attribute)

Is the canonical basis of the quantized universal enveloping algebraU. When this is constructed
nothing is computed. By usingPBWElements (3.10.4), MonomialElements (3.10.5), Strings
(3.10.6) information about elements of the canonical basis can be obtained.

Example
gap> U:= QuantizedUEA( RootSystem( "F", 4 ) );;
gap> B:= CanonicalBasis( U );
<canonical basis of QuantumUEA( <root system of type F4>, Qpar = q ) >

3.10.4 PBWElements

♦ PBWElements( B, rt ) (operation)

HereB is the canonical basis of a quantized uea, andrt a list of non-negative integers representing
an element of the root lattice (e.g., if the simple roots areα, β andrt = [ 3, 2 ], thenrt represents
3α+2β).

It is possible to add the optionlowrank, as followsPBWElements( B, rt :lowrank ). In that
case a somewhat different method will be used, that is significantly faster if the underlying root system
has rank 2,3. It is about equally fast for ranks 4,5; and slower for ranks greater than 5.

Example
gap> U:= QuantizedUEA( RootSystem( "F", 4 ) );;
gap> B:= CanonicalBasis( U );;
gap> PBWElements( B, [1,2,1,0] );
[ F1*F3ˆ(2)*F9, F1*F3*F7+(qˆ4)*F1*F3ˆ(2)*F9, (qˆ4)*F1*F3ˆ(2)*F9+F2*F3*F9,

(qˆ2)*F1*F3*F7+(qˆ2+qˆ6)*F1*F3ˆ(2)*F9+(qˆ2)*F2*F3*F9+F2*F7,
(qˆ4)*F1*F3*F7+(qˆ8)*F1*F3ˆ(2)*F9+(qˆ4)*F2*F3*F9+(qˆ2)*F2*F7+F3*F4 ]

gap> U:= QuantizedUEA( RootSystem("G",2) );;
gap> B:= CanonicalBasis( U );;
gap> PBWElements( B, [2,3] : lowrank );
[ F1ˆ(2)*F6ˆ(3), F1*F5*F6ˆ(2)+(qˆ8+qˆ10)*F1ˆ(2)*F6ˆ(3),

(qˆ2)*F1*F5*F6ˆ(2)+(qˆ6+qˆ12)*F1ˆ(2)*F6ˆ(3)+F3*F6ˆ(2),
(qˆ8)*F1*F5*F6ˆ(2)+(qˆ18)*F1ˆ(2)*F6ˆ(3)+(qˆ6)*F3*F6ˆ(2)+F5ˆ(2)*F6 ]

3.10.5 MonomialElements

♦ MonomialElements( B, rt ) (operation)

This does the same asPBWElements (3.10.4), except that the elements are written as linear com-
binations of monomials in the generatorsFα, whereα runs through the simple roots.

We remark that this information is also computed “behind the scenes” when calling
PBWElements( B, rt ). However, it is not computed if the optionlowrank is present in the call to
PBWElements.

Example
gap> U:= QuantizedUEA( RootSystem( "F", 4 ) );;
gap> B:= CanonicalBasis( U );;
gap> MonomialElements( B, [1,2,1,0] );
[ F1*F3ˆ(2)*F9, F1*F3*F9*F3+(-1)*F1*F3ˆ(2)*F9, F3ˆ(2)*F1*F9, F3*F1*F9*F3,

F3*F9*F3*F1+(-1)*F3ˆ(2)*F1*F9 ]

3.10.6 Strings

♦ Strings( B, rt ) (operation)

HereB, rt are the same as inPBWElements (3.10.4). This returns the list of strings corresponding
to the elements ofB of weightrt (cf. Section2.6). For example, if on thek-th position of the list
returned by this function we have[ 1, 2, 2, 3 ], then the principal monomial of thek-th element
of PBWElements( B, rt ) is F̃2

1 F̃3
2 (1) (whereF̃i is thei-th Kashiwara operator).

We remark that this information is also computed “behind the scenes” when calling
PBWElements( B, rt ). However, it is not computed if the optionlowrank is present in the call to
PBWElements.

Example
gap> U:= QuantizedUEA( RootSystem( "F", 4 ) );;
gap> B:= CanonicalBasis( U );;
gap> Strings( B, [1,2,1,0] );
[ [ 1, 1, 2, 2, 3, 1 ], [ 1, 1, 2, 1, 3, 1, 2, 1 ], [ 2, 2, 1, 1, 3, 1 ],

[ 2, 1, 1, 1, 3, 1, 2, 1 ], [ 2, 1, 3, 1, 2, 1, 1, 1 ] ]
gap> Falpha( Falpha( Falpha( Falpha( One(U), 3 ), 1 ), 2 ), 2 );
F2*F3*F9
gap> PBWElements( B, [1,2,1,0] )[3];
(qˆ4)*F1*F3ˆ(2)*F9+F2*F3*F9

3.10.7 PrincipalMonomial

♦ PrincipalMonomial( u ) (operation)

Hereu is an element of the output ofPBWElements (3.10.4). This function returns the unique
monomial ofu that has coefficient 1.

Example
gap> U:= QuantizedUEA( RootSystem("G",2) );;
gap> B:= CanonicalBasis( U );;
gap> p:= PBWElements( B, [4,4] : lowrank )[4];
(qˆ9)*F1ˆ(2)*F3*F6ˆ(3)+F1ˆ(2)*F5ˆ(2)*F6ˆ(2)+(qˆ9+qˆ11+qˆ13)*F1ˆ(3)*F5*F6ˆ(
3)+(qˆ20+qˆ22+2*qˆ24+qˆ26+qˆ28)*F1ˆ(4)*F6ˆ(4)
gap> PrincipalMonomial( p );
F1ˆ(2)*F5ˆ(2)*F6ˆ(2)

3.10.8 StringMonomial

♦ StringMonomial( u ) (operation)

Hereu is a monomial in the negative part of a quantized enveloping algebra, e.g., as output by
PrincipalMonomial (3.10.7). This function computes the corresponding “string” (see Section2.6).
The strings are output in the same way as in3.10.6.

Example
gap> U:= QuantizedUEA( RootSystem("G",2) );;
gap> B:= CanonicalBasis( U );;
gap> p:= PBWElements( B, [1,2] : lowrank )[2];;
gap> m:=PrincipalMonomial( p );
F5*F6
gap> StringMonomial( m );
[ 2, 2, 1, 1 ]
gap> Falpha( Falpha( Falpha( One(U), 1 ), 2 ), 2 );
F5*F6

3.10.9 Falpha (for a module element)

♦ Falpha( V, v, ind ) (operation)

HereV is a module over a quantized enveloping algebra,v an element of it, andind an index
between 1 and the rank of the root system. The function returns the result of applying theind-th
Kashiwara operator̃Find to v. Here the Kashiwara operators are different from the ones described in
Section2.6. We refer to [Jan96], 9.2 for the definition of the operators used here.

Example
gap> U:= QuantizedUEA( RootSystem("B",2) );;
gap> V:= HighestWeightModule( U, [1,1] );;
gap> Falpha( V, Basis(V)[1], 1 );
F1*v0

3.10.10 Ealpha (for a module element)

♦ Ealpha( V, v, ind ) (operation)

HereV is a module over a quantized enveloping algebra,v an element of it, andind an index
between 1 and the rank of the root system. The function returns the result of applying theind-th
Kashiwara operator̃Eind to v. Here the Kashiwara operators are different from the ones described in
Section2.6. We refer to [Jan96], 9.2 for the definition of the operators used here.

Example
gap> U:= QuantizedUEA( RootSystem("B",2) );;
gap> V:= HighestWeightModule( U, [1,1] );;
gap> v:= Falpha( V, Basis(V)[2], 2 );
(qˆ2)*F1*F4*v0+F2*v0
gap> Ealpha( V, v, 2 );
F1*v0

3.10.11 CrystalBasis

♦ CrystalBasis( V ) (attribute)

HereV is a finite-dimensional left module over a quantized enveloping algebra. This function
returns the canonical, or crystal basis of V (see Section2.6).

This function only works for modules for whichWeightsAndVectors (3.8.14) works.
Example

gap> U:= QuantizedUEA( RootSystem( "B", 2 ) );;
gap> V:= HighestWeightModule( U, [1,1] );
<16-dimensional left-module over QuantumUEA( <root system of type B2>, Qpar
= q )>
gap> CrystalBasis( V );
Basis( <16-dimensional left-module over QuantumUEA( <root system of type B
2>, Qpar = q )>, [ 1*v0, F1*v0, F4*v0, F1*F4*v0, (qˆ2)*F1*F4*v0+F2*v0, F2*F4*v0,
(q)*F2*F4*v0+F3*v0, (-qˆ-4)*F1*F2*v0, (-qˆ-1)*F1*F3*v0+(-qˆ-3)*F2ˆ(2)*v0,
(-qˆ-2)*F2ˆ(2)*v0, F3*F4*v0, (-qˆ-4)*F2*F3*v0+(-qˆ-2)*F2ˆ(2)*F4*v0,
(-qˆ-2)*F2*F3*v0, (qˆ-4)*F2ˆ(3)*v0, (-qˆ-1)*F3ˆ(2)*v0, (qˆ-5)*F2ˆ(2)*F3*v0 ] )

3.10.12 CrystalVectors

♦ CrystalVectors( V ) (attribute)

HereV is a finite-dimensional left module over a quantized enveloping algebra. LetC be the crystal
basis ofV (i.e., output byCrystalBasis (3.10.11)). This function returns a list of cosets of the basis
elements ofC moduloqL, whereL is theZ[q]-lattice spanned byC.

The coset of a vectorv is printed as<v>.
The crystal vectors are used to construct the point set of the crystal graph ofV (CrystalGraph:for

a module (3.10.15)).
This function only works for modules for whichWeightsAndVectors (3.8.14) works.

Example
gap> U:= QuantizedUEA( RootSystem( "B", 2 ) );;
gap> V:= HighestWeightModule( U, [1,1] );
<16-dimensional left-module over QuantumUEA( <root system of type B

2>, Qpar = q )>
gap> CrystalVectors( V );
[ <1*v0>, <F1*v0>, <F4*v0>, <F2*v0>, <F1*F4*v0>, <F3*v0>,

<(-qˆ-4)*F1*F2*v0>, <F2*F4*v0>, <F1*F3*v0>, <F3*F4*v0>,
<(-qˆ-1)*F1*F3*v0+(-qˆ-3)*F2ˆ(2)*v0>, <(-qˆ-4)*F2*F3*v0+(-qˆ-2)*F2ˆ(2)*F

4*v0>, <F2ˆ(2)*F4*v0>, <(qˆ-4)*F2ˆ(3)*v0>, <(-qˆ-1)*F3ˆ(2)*v0>,
<(qˆ-5)*F2ˆ(2)*F3*v0> ]

3.10.13 Falpha (for a crystal vector)

♦ Falpha( v, ind ) (operation)

Herev is a crystal vector, i.e., an element ofCrystalVectors( V ), whereV is a left module
over a quantized enveloping algebra. This function returns the result of applying theind-th Kashiwara
operatorF̃αind

to v. The result isfail if F̃αind
(v)=0.

Example
gap> U:= QuantizedUEA( RootSystem( "B", 2 ) );;
gap> V:= HighestWeightModule( U, [1,1] );;
gap> c:=CrystalVectors( V );;
gap> Falpha( c[2], 2 );
<F2*v0>
gap> Falpha( c[3], 2 );
fail
gap> Falpha( Falpha( Falpha( c[1], 1 ), 2 ), 1 );
fail
gap> p:= DominantLSPath( RootSystem( "B", 2 ), [1,1] );
<LS path of shape [ 1, 1 ] ending in [ 1, 1 ] >
gap> Falpha( Falpha( Falpha( p, 1 ), 2 ), 1 );
fail

The last part of this example is an illustration of the fact that the crystal graph of a highest-weight
module can be obtained by the path method (see Section2.7).

3.10.14 Ealpha (for a crystal vector)

♦ Ealpha( v, ind ) (operation)

Herev is a crystal vector, i.e., an element ofCrystalVectors( V ), whereV is a left module
over a quantized enveloping algebra. This function returns the result of applying theind-th Kashiwara
operatorẼαind

to v. The result isfail if Ẽαind
(v)=0.

Example
gap> U:= QuantizedUEA( RootSystem( "B", 2 ) );;
gap> V:= HighestWeightModule( U, [1,1] );;
gap> c:=CrystalVectors( V );;
gap> Ealpha( c[3], 1 );
fail
gap> Ealpha( c[3], 2 );
<1*v0>

3.10.15 CrystalGraph (for a module)

♦ CrystalGraph( V ) (function)

Returns the crystal graph of the moduleV. The points of this graph are the cosets output by
CrystalVectors (3.10.12). The edges work in the same way as inCrystalGraph:for root
system and weight (3.9.7).

Example
gap> U:= QuantizedUEA( RootSystem("A",2) );;
gap> V1:= HighestWeightModule( U, [1,0] );;
gap> V2:= HighestWeightModule( U, [0,1] );;
gap> W:= TensorProductOfAlgebraModules( V1, V2 );;
gap> CrystalGraph( W );
rec(

points := [ <1*(1*v0<x>1*v0)>, <1*(F1*v0<x>1*v0)>, <1*(1*v0<x>F3*v0)>,
<1*(1*v0<x>F2*v0)+qˆ-1*(F2*v0<x>1*v0)>,
<-qˆ-1*(1*v0<x>F2*v0)+qˆ-1*(F1*v0<x>F3*v0)>, <1*(F2*v0<x>F3*v0)>,
<-qˆ-1*(F1*v0<x>F2*v0)>, <-qˆ-1*(F2*v0<x>F2*v0)>,
<-qˆ-3*(1*v0<x>F2*v0)+-qˆ-1*(F1*v0<x>F3*v0)+1*(F2*v0<x>1*v0)> ],

edges := [ [ [ 1, 2 ], 1 ], [ [ 1, 3 ], 2 ], [ [ 2, 4 ], 2 ],
[ [ 3, 5 ], 1 ], [ [ 4, 6 ], 2 ], [ [ 5, 7 ], 1 ], [ [ 6, 8 ], 1 ],
[ [ 7, 8 ], 2 ] ] )

3.11 Universal enveloping algebras

Here we describe functions for connecting a quantized enveloping algebra to the corresponding uni-
versal enveloping algebra.

3.11.1 UEA

♦ UEA( L ) (attribute)

This function returns the universal enveloping algebrau of the semisimple Lie algebraL. The gen-
erators ofu are the generators of a Kostant lattice in the universal enveloping algebra (these generators
are obtained fromL by LatticeGeneratorsInUEA( L ), see theGAP reference manual).

Example
gap> L:= SimpleLieAlgebra( "B", 2, Rationals );
<Lie algebra of dimension 10 over Rationals>
gap> u:= UEA( L );
<algebra over Rationals, with 10 generators>
gap> g:= GeneratorsOfAlgebra( u );
[ y1, y2, y3, y4, x1, x2, x3, x4, ( h9/1 ), ( h10/1 ) ]

3.11.2 UnderlyingLieAlgebra

♦ UnderlyingLieAlgebra( u ) (attribute)

For a universal enveloping algebrau constructed byUEA (3.11.1), this returns the corresponding
semisimple Lie algebra

Example
gap> L:= SimpleLieAlgebra( "B", 2, Rationals );;
gap> u:= UEA( L );;
gap> UnderlyingLieAlgebra( u );
<Lie algebra of dimension 10 over Rationals>

3.11.3 HighestWeightModule (for a universal env. alg)

♦ HighestWeightModule( u, hw ) (operation)

For a universal enveloping algebrau constructed byUEA (3.11.1), this returns the irreducible high-
est weight module overu with highest weighthw, which must be dominant. This module is the same
as the corresponding highest weight module over the semisimple Lie algebra, but in this case the
enveloping algebrau acts.

Example
gap> L:= SimpleLieAlgebra( "B", 2, Rationals );;
gap> u:= UEA( L );;
gap> HighestWeightModule( u, [2,3] );
<140-dimensional left-module over <algebra over Rationals, with
10 generators>>

3.11.4 QUEAToUEAMap

♦ QUEAToUEAMap( L ) (attribute)

HereL is a semisimple Lie algebra. Setu := UEA( L ), andU := QuantizedUEA( L ) (so
u, U are the universal enveloping algebra, and “generic” quantized enveloping algebra ofL respec-
tively). ThenQUEAToUEAMap( L ) returns the algebra homomorphism fromU to u obtained by map-
ping q to 1, a generatorFi, corresponding to a simple root to the generatoryi (corresponding to the
same simple root), and likewise forEi andxi. This means thatKi is mapped to one, and[ Ki : s ]
to hi chooses.

The canonical basis ofU is mapped to the canonical basis ofu.
Example

gap> L:= SimpleLieAlgebra( "B", 2, Rationals );;
gap> f:= QUEAToUEAMap( L );
<mapping: QuantumUEA( <root system of rank
2>, Qpar = q ) -> Algebra( Rationals, [ y1, y2, y3, y4, x1, x2, x3, x4,

( h9/1 ), ( h10/1 ) ] ) >
gap> U:= Source( f );
QuantumUEA( <root system of rank 2>, Qpar = q )
gap> u:= Range( f );
<algebra over Rationals, with 10 generators>
gap> B:= CanonicalBasis( U );;
gap> p:= PBWElements( B, [1,2] );
[ F1*F4ˆ(2), (q+qˆ3)*F1*F4ˆ(2)+F2*F4, (qˆ4)*F1*F4ˆ(2)+(q)*F2*F4+F3 ]
gap> pu:= List( p, x -> Image( f, x ) );
[ y1*y2ˆ(2), 2*y1*y2ˆ(2)+y2*y3-2*y4, y1*y2ˆ(2)+y2*y3-1*y4 ]
gap> V:= HighestWeightModule( u, [2,1] );
<40-dimensional left-module over <algebra over Rationals, with
10 generators>>

gap> List( pu, x -> xˆBasis(V)[1] );
[ 0*v0, y2*y3*v0+-2*y4*v0, y2*y3*v0+-1*y4*v0 ]
# Which gives us a piece of the canonical basis of V.

